File handling

It’s a funny thing, this Delphi. Before it came along, people who bought a Pascal compiler would probably learn how to talk to normal files using standard file management routines. If they wanted to talk to established database formats they would need to obtain an appropriate add-on library, such as the Paradox Engine or the Borland Database Engine, and negotiate the many APIs it required in particular orders with the appropriate parameter sets. Delphi has made database access of practically any description a breeze. So much has it turned the tables (pun unintended) that many Delphi users wouldn’t know where to look for file-management routines not centred around a database table.

This section will explore what Delphi has to offer on the subject of file manipulation. The principle benefactors will be those readers new to file handling, but included should be enough bits and bobs so that even the experienced will pick up one or two things along the way. Subjects to be covered include basic file access using file variables (text, typed and untyped), file handles, one or two efficiency tips, file sharing, record locking (and no I’m not sneakily introducing a database section), text file device drivers and streams.

So what do we need to know how to do, with respect to files? Things we may wish for include creating files, writing data to files, both sequentially and in a random access fashion, reading from files (again sequentially and random access), renaming files, copying files, deleting files, making directories, finding files and probably more besides. Fortunately, this product has copious subroutines available to do all these things and more.

File variables

First things first, though - how do we represent a file in a program? Pascal offers two ways, if we ignore objects for the time being: the historic Pascal file variable and the more recent file handle, corresponding to the options C programmers have. We’ll deal with file variables first and come back to file handles later. There are three forms of file variable, and the type we use depends on what form of file we wish to address. If we are interested in a text file, we use the TextFile type, defined in the System unit (the unit that is implicitly used by every unit and project source file, i.e. the unit that implements the core run-time library). TextFile is a substitute for Text, which was available in Borland’s previous Pascal compilers, and has been introduced to prevent ambiguity with the various Text properties in the VCL. It is often still safe to use Text, but if you wish to use it, it is best to fully qualify it, i.e. use System.Text. Suffixing File (TextFile) is more Delphi-like, but prefixing System. (System.Text) will retain compatibility with Borland Pascal.

If we are looking at a non-text file we need to identify if there is a regular record structure to the file. If so, we can use a typed file variable, otherwise an untyped file variable. Some example variable definitions follow, where ATextFile is clearly intended for use with a text file. AnUntypedFile can be used for any arbitrary file and ATypedFile can be used for a file which is made up of Double values. You can see that a typed file definition is an extension of an untyped file, specifying the type of data constituting the file.

var

  { Could use System.Text instead of TextFile }

  ATextFile: TextFile;

  AnUntypedFile: File;

  ATypedFile: File of Double;

A file variable is not much use until it has been associated with a real file name - this is done with the AssignFile procedure. Again, this is a new routine in Delphi Pascal and is used in place of Assign to avoid scoping problems with the Assign methods of various components. A file assignment looks like this:

{ Can use System.Assign over AssignFile }

AssignFile(ATextFile, ‘C:\AUTOEXEC.BAT’);

Now we are ready to run - we can open the file. How we do this depends on whether we want to read or write the file, or both. A text file can be opened for reading or writing; typed and untyped files can be opened for reading, writing or both. The Rewrite procedure will create a new file and open it for writing, for any file type (though see below). Append can be used to open an existing text file so more text can be added at the end. The Reset procedure operates differently depending on the file type. If given a text file, Reset opens the file read-only, but if it is given a typed or untyped file, it opens it in read/write mode (by default - we’ll come back to how to change this later in the File Sharing section) to allow random access reading and writing. Reset and Rewrite are often used in combination to open an existing file, ensuring it exists first:

{ Create empty file if none found }

if not FileExists(FileName) then

  Rewrite(FileVar);

Reset(FileVar); { Open file }

Closing the file is done with CloseFile, a Delphi replacement for the old Close procedure. You should use try..finally blocks if you will be opening and closing a file in any sequence of instructions. Use of these constructs will protect the CloseFile call from being missed out if some exception occurs.

Reset(AnUntypedFile);

try

  ...

finally

  CloseFile(AnUntypedFile); { or System.Close }

end;

Delphi 2 Rewrite bug

Note that Delphi 2.00 (not 2.01) exhibits erroneous behaviour with successive operations on a file. If an AssignFile call is made followed by a Rewrite, then a new file is created. If the file is closed and Reset is called, it is opened for reading. If the file is closed again and Rewrite called again, the file should be rewritten. In Delphi 1 that is what happens. In Delphi 2 the file is opened in append mode, and gets extended rather than overwritten. The code below

To ensure that Rewrite overwrites the existing file (as per the documentation), call AssignFile before every call to Rewrite.

A bug in Delphi 2.00 means that under certain circumstances, Rewrite acts rather like Append. This is fixed in 2.01.

Code breaks in Delphi 2 on second call to Rewrite
Code works fine in Delphi 1 and 2

AssignFile(F, 'c:\deleteme.txt');

Rewrite(F);

try

  WriteLn(F, 'First line');

finally

  CloseFile(F);

end;

Reset(F);

try

  ReadLn(F, S);

finally

  CloseFile(F);

end;

{ Rewrite called after Reset }

Rewrite(F);

try

  WriteLn(F, 'First line');

finally

  CloseFile(F);

end;
AssignFile(F, 'c:\deleteme.txt');

Rewrite(F);

try

  WriteLn(F, 'First line');

finally

  CloseFile(F);

end;

Reset(F);

try

  ReadLn(F, S);

finally

  CloseFile(F);

end;

AssignFile(F, 'c:\deleteme.txt');

Rewrite(F);

try

  WriteLn(F, 'First line');

finally

  CloseFile(F);

end;

Records and buffers

A typed file has a clear concept of records, each component item in the file is termed a record and the record size for the file is the size of each item. For ATypedFile above, the record size is the size of a Double, i.e. SizeOf(Double) or 8 bytes. With an untyped file it is not so clear. An impulse might be to think that an untyped file is simply a collection of bytes and so it has a record size of 1 byte. This is not necessarily the case - if it were a file of bytes, why not insist it is declared as File of Byte? In fact by default, Reset gives an untyped file a record size of 128 bytes. It isn’t quite so important these days with clever disk caching that we all tend to have but historically, reading/writing files 128 bytes at a time is considerably more efficient than reading one byte at a time. If you want to change this default untyped file record size, you can use an optional second parameter to Reset:

Reset(AnUntypedFile, 2048);

Text files

Text files do not have a record size as such, but there is a buffer associated with each text file which also defaults to 128 bytes and it is used to read or write text files 128 bytes at a time for efficiency. Again, you can change this if you wish, this time using the SetTextBuf routine, which you pass a buffer of any type to and this will be used until Assign is next called for the text file variable. Make sure the buffer will be in existence for as long as the text file will be open - bad things will happen if you pass a local variable which goes out of scope before the file is closed. Also, do not call SetTextBuf after accessing the file or you will likely lose data. One last point is to note that all controls have a SetTextBuf method, and so it may be necessary to use a scope override: System.SetTextBuf to remove ambiguity.

var

  ATextFile: TextFile;

  Buffer: array[1..2048] of Char;

procedure DoFileStuff;

begin

  AssignFile(ATextFile, ‘BigFile.Dat’);

  Reset(ATextFile);

  try

    SetTextBuf(ATextFile, Buffer);

    { Use the text file }

    ...

  finally

    CloseFile(ATextFile);

  end;

end;

If you don’t want the file to use all the buffer, you can use an optional third parameter to limit how much it uses. This extra parameter defaults to the size of the passed in buffer.

  SetTextBuf(ATextFile, Buffer,(
    SizeOf(Buffer) div 2);

Since there isn’t really a record size associated with a text file, there’s just a buffer that fills up as you write to it, you can end up with data written to the text file variable sitting in memory for some time (until the file is closed). If you want to ensure the data is sent to the operating system to do with what it will, but don’t feel like closing the file, you can call the Flush routine:

  Flush(ATextFile);

Reading and writing

So that’s some housekeeping out of the way; now we need to find how to access data in the file. To read and write data from text files or typed files, we use the Read and Write routines. These will be very familiar to Borland Pascal users as the way you also read and write to the standard Input/Output devices (keyboard and screen), but perhaps not so to many Delphi users where by default there are no standard I/O devices (though you can set some up as shown later). Read and Write are unusual calls since they can take a variable number of arguments (mind you, so do Reset and SetTextBuf and I didn’t raise an eyebrow at them). It is not possible to define your own routines to do the same, although Delphi does allow it with the penalty of an extra pair of square brackets (e.g. consider the Format function, and the TTable.FindKey method - you can find out how to define your own routines like this in the Variable parameter lists and dynamic array creation section). The first parameter you need to pass is the file variable, and this is followed by the data you wish to write. In the case of a text file this can be integers, floating point numbers etc., each of which get written to the file as their textual equivalent. Additionally, for text files, you can use ReadLn and WriteLn to read or write whatever values, if any, are passed followed by a carriage return and line feed. For typed files, the values must match the component type of the file.

var

  ATextFile: TextFile;

  ATypedFile: File of Double;

procedure DoMoreFileStuff(Dbl: Double);

begin

  WriteLn(ATextFile, 1, ‘ Hello ’, 5.5);

  Write(ATypedFile, Dbl);

end;

Untyped files can’t use those routines; instead they must use BlockRead and BlockWrite.

Note that the third and fourth parameters of BlockRead and BlockWrite are both Words in Delphi 1 (0..65535), but are Integers in Delphi 2 (-2Gb..2Gb-1). This change is to cater for the 32-bit file system which deals in 32-bit values. In the program they are defined as Cardinal, so they will be 16-bit in Delphi 1 and 32-bit in Delphi 2.

To see the extent of file variable support in the libraries, the tables below respectively list all the routines that work solely with text file variables, those that work with typed and untyped file variables, the untyped file variable routines and lastly those that work with any file variable, regardless of type.

System unit (Delphi and Borland Pascal)

Append
opens a text file in write-only mode allowing text to be added at the end

Eoln
returns True if the file pointer is at the end of a line

Flush
flushes text file’s internal 128-byte (by default) buffer to DOS

ReadLn
reads a line of a text file

SeekEof
like Eof but skips past white spaces and end of lines before performing the test

SeekEoln
same as Eoln but skips past white spaces before performing the test

SetTextBuf
replaces the 128 byte I/O buffer with a different (usually bigger one) for more efficient text file operations

WriteLn
writes a line of text at the current file pointer

File variable routines for text files only

System unit (Delphi and Borland Pascal)

FilePos
returns position of file pointer of an open file variable in terms of records

FileSize
returns file size in terms of records (typed file) or bytes (untyped file)

Seek
moves the file pointer to a particular record

Truncate
sets the current file position as the end of file

File variable routines for typed files and untyped files

System unit (Delphi and Borland Pascal)

BlockRead
reads arbitrary number of records from a file

BlockWrite
writes an arbitrary number of records to a file

File variable routines for untyped files only

System unit (Delphi and Borland Pascal)

Assign
associates a file variable with an file name

AssignFile
Delphi substitute for Assign to avoid scoping problems

Close
closes a file

CloseFile
Delphi substitute for Close to avoid scoping problems

Eof
returns True if file pointer is at the end of the file

Erase
deletes a file

Read
reads data from a file

Rename
renames a file

Reset
opens a file. Text files are read-only, other files are read/write but can be changed with the FileMode variable

Rewrite
creates and opens a new file

Write
writes data to a file

WinDos unit (Delphi 1 and Borland Pascal) and Dos unit (Borland Pascal only)

GetFAttr
finds a file’s attributes. Delphi introduces FileGetAttr

GetFTime
used with UnpackTime to find file’s last modification time and date. Delphi introduces FileGetDate

SetFAttr
sets a file’s attributes. Delphi introduces FileSetAttr

SetFTime
used with PackTime to set file’s last modification time and date. Delphi introduces FileSetDate

File variable routines for any files

Okay, let’s see some code, firstly for text files. Here’s a routine that dumps a text file to the screen by reading lines from the file until the end of file is encountered. It’s used by a project called DUMPTEXT.DPR which writes your AUTOEXEC.BAT file to the screen.

procedure DumpTextFile(const FileName: String);

var

  F: TextFile;

  S: String;

begin

  try

    AssignFile(F, FileName);

    Reset(F);

    while not EOF(F) do

    begin

      ReadLn(F, S);

      WriteLn(S);

    end;

  finally

    CloseFile(F);

  end;

end;

Note that in a GUI Windows program there are no standard input or output devices to get data from or present information to the user by default. In Delphi 1.0x we can use the WinCrt unit to emulate them. In Delphi 2.0 we can generate console mode applications (which use windows just like DOS programs do, i.e. either text mode or windowed text) by setting a linker option (Project | Options | Linker | EXE and DLL options | Generate console application), or using the {$AppType Console} compiler directive (if you are unfamiliar with these, refer to the Compiler directive section).

Then, whenever a Write(Ln) or Read(Ln) instruction appears without a file variable the standard input (keyboard) will be read from or the standard output (CRT emulation or text window) will be written to. Without WinCrt, such an instruction in a 16-bit error will give run-time error 104 (file not open for input) or 105 (file not open for output). In a 32-bit program you get error 104 (file not open for input) or 103 (file not open). The statements:

WriteLn(‘Hello’);

ReadLn(StringVar);

are treated as being:

WriteLn(Output, ‘Hello’);

ReadLn(Input, StringVar);

where Input and Output are two predefined text file variables. Choosing a WinCrt or console application associates these files with some device to allow them to work.

The previous routine dealt with lines as individual entities, but often this will not be a sensible approach. Consider a tab separated data file, with 5 entries per line interspersed with tab characters. A better way to treat such a file is to use Read instead of ReadLn, and check for the end of the line with Eoln or SeekEoln. To show the difference between these two functions, try this routine (from the program READDATA.DPR) which starts off using Eoln.

procedure WriteAndReadDataFile;

var

  F: TextFile;

  Loop, Num: Byte;

begin

  AssignFile(F, 'C:\DELETEME.DAT');

  try

    Rewrite(F);

    for Loop := 1 to 3 do

      Write(F, Loop, #9);

    WriteLn(F);

    Reset(F);

    { Change Eoln to SeekEoln to work properly }

    while not Eoln(F) do

    begin

      Read(F, Num);

      WriteLn(Num);

    end;

  finally

    CloseFile(F);

  end;

end;

Notice the numbers are written out to the first line of the file with tab separators (ASCII value 9), and when all have been written, a line terminator is written by calling WriteLn(F). The numbers 1, 2 and 3 are written to the file, but the numbers read back and written to the screen are 1, 2, 3 and 0.



#####READ1.BMP

The problem here is that after the 3, another tab character is written out. During the read loop, when the 3 has been read back in, the end of the line has not been reached - the tab character remains, and so another iteration of the loop takes place. No number exists after the three, so the Read returns a zero. In short, the extra white space mucked things up. This is where SeekEoln comes in. SeekEoln eats up any white space before deciding whether it is at the end of the line. Changing Eoln to SeekEoln changes the program to do what we would expect.



#####READ2.BMP

Copying files

Typically, Pascal users have employed BlockRead and BlockWrite to copy files around disk drives, operating against untyped file variables. One possible implementation is found in CopyU1.Pas and used in CopyEg1.Dpr.

procedure FileCopy(const InFileName, OutFileName: String);

const

  BufSize = 8 * 4096; { 32kb }

type

  PBuffer = ^TBuffer;

  TBuffer = array[1..BufSize] of Byte;

var

  Size: Cardinal;

  Buffer: PBuffer;

  InFile, OutFile: File;

begin

  if InFileName = OutFileName then

    raise EInOutError.Create('File cannot be copied onto itself')

  else

  begin

    Buffer := nil;

    Assign(InFile, InFileName);

    Reset(InFile, 1);

    try

      Assign(OutFile, OutFileName);

      Rewrite(OutFile, 1);

      try

        New(Buffer);

        repeat

          BlockRead(InFile, Buffer^, BufSize, Size);

          BlockWrite(OutFile, Buffer^, Size)

        until Size < BufSize;

        FileSetDate(TFileRec(OutFile).Handle,

          FileGetDate(TFileRec(InFile).Handle));

      finally

        if Buffer <> nil then

          Dispose(Buffer);

        CloseFile(OutFile)

      end;

    finally

      CloseFile(InFile);

    end

  end

end;

It uses a 32k buffer allocated from the heap with New, and deallocated with Dispose. Note the use of try..finally blocks to ensure the files get closed and the heap deallocated in the event of an exception. Also note the calls to FileSetDate and FileGetDate used in conjunction with some file variable typecasts to access the file handle from a file variable. This technique will be looked at again later in the section called File variable structure records.

Catering for a progress indicator

A reasonably simple modification can be made to this routine (see CopyU2.Pas for the implementation and CopyEg2.Dpr for an example of using it) to get feedback on long copies. The code below shows the routine being called, passing in a callback routine which will be called periodically through a lengthy copy. Any callback can be passed in, providing it is a procedure and takes the same parameter types.

Note that this example uses a Windows 95 progress bar and so can only be run in Delphi 2. There is another example, CopyEg2b.Dpr which uses a Delphi 1 gauge component and can be used in either version.

procedure FileCopyCallBack(Position, Size: LongInt); far;

var

  Factor: Longint;

begin

  { Progress bars have a max. value of 64k-1 }

  { Scale our values down if necessary }

  Factor := Size div High(TProgressRange);

  if Factor > 1 then

  begin

    Size := Size div High(TProgressRange);

    Position := Position div High(TProgressRange);

  end;

  Form1.ProgressBar.Max := Size;

  Form1.ProgressBar.Position := Position;

end;

procedure TForm1.BtnCopyClick(Sender: TObject);

begin

  BtnCopy.Enabled := False;

  FileCopy(EdtCopyFrom.Text, EdtCopyTo.Text, FileCopyCallBack);

  BtnCopy.Enabled := True;

end;

Remember that in Delphi 1 procedures assigned to procedural variables (as this callback is) must be compiled in the far call model. The easiest way to ensure this is to add the word far onto the first occurrence of the procedure header. The listing above does this despite the fact that it is Delphi 2 specific. Delphi 2 ignores the far directive – it has no meaning in 32-bit.

Untyped files

Now for a look at untyped file variables. The example I’m using here operates on a Windows 3.1x executable file. Pietrek (1994) tell us that Windows 95 can be of particular help for the badly behaved applications that plagued Windows 3.1. In the executable file header is a number indicating the expected Windows version. Windows 3.1 applications use a word where the high byte is 3 and the low byte is 10, indicating 3.10. If this is changed to 4.00, Windows 95 will tidy up Windows resources left hanging around by sloppy programming. A Windows executable is a good example of a file with no regular structure - there are various tables and values, as partly documented in the Windows API help file available in the Delphi 1 help, in the topic Executable-File Header Format.

The routine below is used in the program CHANGAPP.DPR. This program takes a command-line parameter (you can set one up in the development environment by selecting Run | Parameters...) which should be a Windows 3.1x executable file. It changes the expected version to 4.0.

type

  TWordRec = record

    Lo, Hi: Byte;

  end;

procedure UpdateApp(const FileName: String);

var

  F: File;

  Num, NewExeOffset: Word;

const

  Sig: String[2] = '  ';

begin

  AssignFile(F, FileName);

  try

    Reset(F, 1);

    BlockRead(F, Sig[1], SizeOf(Word));

    if Sig <> 'MZ' then

    begin

      WriteLn('Not a valid EXE file');

      Exit;

    end;

    Seek(F, $18);

    BlockRead(F, Num, SizeOf(Word));

    if Num < $40 then

    begin

      WriteLn('Not a Windows EXE file');

      Exit;

    end;

    Seek(F, $3C);

    BlockRead(F, NewExeOffset, SizeOf(Word));

    Seek(F, NewExeOffset);

    BlockRead(F, Sig[1], SizeOf(Word));

    if Sig <> 'NE' then

    begin

      WriteLn('Not a Windows EXE file');

      Exit;

    end;

    Seek(F, NewExeOffset + $3E);

    BlockRead(F, Num, SizeOf(Word));

    with TWordRec(Num) do

      WriteLn('Current expected Windows version: ',(
        Hi, '.', Lo);

    WriteLn('Setting to 4.0');

    Seek(F, NewExeOffset + $3E);

    Num := $400;

    BlockWrite(F, Num, SizeOf(Word));

  finally

    CloseFile(F);

  end;

end;

As it turns out, all values the program read and writes are word-sized, but not all of them are Words. The program opens the file with a record size of 1, so that the offsets in the .EXE file format documentation can be used readily. The first thing it does is check the first two bytes are MZ which implies it is an executable file. BlockRead is used to read them into a two byte string to facilitate this comparison. Then the word at offset $18 is tested against $40. If the value is greater or equal, it means the value at offset $3C represents the offset in the file of the Windows header, and that header is tested to ensure it starts with NE.

$3E bytes past the start of the Windows header is the expected Windows version - this is read in and written out to the screen. A typecast is used to allow the high and low bytes to be written individually. An alternative and perhaps more readable version of that particular section would have involved reading two individual byte values, since that is what is being written out, but never mind. Isn’t hindsight is a wonderful thing? If you are unfamiliar with the principles of typecasting, see the Typecasting section.

To prove that non-text files can be used as random access read/write devices, the file pointer is moved back to the place where the version is stored and a new version is written with BlockWrite, and finally the file is closed.

You may have noticed that I defined a record TWordRec to access the two bytes of a word with, in the context of a typecast. Whilst there is a perfectly suitable WordRec type in the SysUtils unit, I chose not to use that unit, as it carries a lot of baggage that will be added into my .EXE by of which I will be making no use - date and time formatting information, hardware exception handling etc. This information defeats the smart linker as can be seen by adding SysUtils into the uses statement, recompiling and comparing .EXE sizes.

Typed files

Text files and untyped files we can tick off now. Onwards to typed files. A typed file is essentially a binary data file that happens to have some order or structure. Let’s consider that we might like to store numbers from zero to 90 along with their sines, cosines and tangents in a file to save calculating them all the time in a program that has heavy trigonometry usage. Well, yes I know, I’m stretching reality a bit here - it’s hardly likely to be more efficient what with all the required file access, but it will suffice for the purposes of our demonstration. We can define a record to hold the information as:

type

  TMaths = record

    Val: Byte;

    Sine, Cosine, Tangent: Double;

  end;

A routine to set up the file could look like this (see MATHS.DPR for a program that uses it).

procedure SetupFile;

var

  Maths: TMaths;

  Loop: Byte;

  F: File of TMaths;

begin

  AssignFile(F, 'C:\DELETEME.DAT');

  try

    Rewrite(F);

    for Loop := 1 to 90 do

    begin

      with Maths do

      begin

        Val := Loop;

        Sine := Sin(Loop);

        Cosine := Cos(Loop);

        Tangent := Sine/Cosine;

      end;

      Write(F, Maths);

    end;

  finally

    CloseFile(F);

  end;

end;

If you run the program and look at the resultant file, you will see that it is an unreadable binary file. If the file is opened with Reset, it will be a random access read/write file, just like the previous untyped file example. The Seek routine will seek to any individual record in the file and read or write it. If we go to the end of the file, possibly by calling Seek(FileSize(F)) and write new values, the file will be extended.

File variable structure records

Given any file variable, you can find various pieces of information, such as the record size, the file handle, and whether it is open or closed. This is not obvious as the type file of xxxx looks fairly atomic, with no data fields. However, internally a file variable of type TextFile is stored as a TTextRec and other file variables are stored as TFileRec. These records are defined in the SysUtils unit and can be used to typecast file variables into more interesting data structures. For example, to get the file handle of a given file variable, refers to TFileRec(FileVar).Handle. The record size is given by TFileRec(FileVar).RecSize. You can find what state a file is in by looking at the TFileRec(FileVar).Mode field. It will contain one of four values (again from SysUtils).

SysUtils constant
Value
Meaning

fmClosed
$D7B0
File is closed

fmInput
$D7B1
File is open for reading

fmOutput
$D7B2
File is opened for writing

fmInOut
$D7B3
File is opened for reading and writing

Note that the Delphi on-line help suggests using these values for opening and closing disk files. Don’t. Use these only to find the state of a file. The TTextRec type is explored a bit further in the later Text File Device Drivers part of this File Handling section

Error handling

File variable error handling can be done in two ways - either with exception handling (the Delphi way) or without (the Turbo Pascal way). Which approach you use depends on the state of the I/O checking option in your code. I/O checking can be turned on or off in the Runtime errors section of the project options dialog, on the Compiler page. This setting acts on a global basis, affecting all source code compiled after that time, or on a more localised basis using the {$I+} and {$I-} compiler directives.

Exceptions and Run-Time Errors

If I/O checking is enabled, the compiler inserts a call to a routine that checks for I/O errors after every file variable I/O call. If SysUtils has been used somewhere in your project, then any I/O error will generate a run-time error, which is turned into an EInOutError exception (which has a public Integer data field called ErrorCode whose possible values are listed in the large table below), otherwise it will cause a terminal run-time error. The run-time error codes are the same as exception error codes and so Table 1 also applies. If you use SysUtils and for some reason want to remove run-time error exceptions from your program, you can assign a new value to ErrorProc, e.g.

ErrorProc := nil;

This statement means that all software run-time errors (out of memory, range check error etc., but not GPFs, floating point overflow etc.) will display an unattractive system modal message box before the program exits. To avoid this, but still be left without the normal run-time error exceptions, we can set up a custom run-time error handler that brings up a custom message, maybe does some tidying up, and then exits. The approach to do this is undocumented, but scouring the source to the SysUtils unit, and some of the assembler source of the System unit reveals the secrets. We need a routine to assign to ErrorProc that does an appropriate job. See later for more details on this subject.

This is an example code snippet using exception handling to trap I/O errors. Run-time errors can not normally be trapped - they terminate the application

AssignFile(F, 'data.dat');

try

  Reset(F);

  try { All's well }

    ...

  finally

    CloseFile(F);

  end;

except { All's not well }

  on E: EInOutError do

    case E.ErrorCode of

      1..6, 15, 100..106: { Oops };

    else

      ;

    end;

end;

IOResult

If I/O checking is disabled, you need to call the System function IOResult after each I/O call. If an I/O call fails, a flag is set preventing further I/O operations succeeding. IOResult returns a value indicating the problem (see the table below) and clears the flag. If you want to get direct access to the flag, you can talk to the InOutRes variable. The flag can be manually cleared by assigning zero to it.

This listing shows an example of an error handler coded using IOResult. Notice that if IOResult is zero, the I/O operation (Reset in this case) succeeded, and so a try..finally block is used to ensure the file is closed after being used.

AssignFile(F, 'data.dat');

{$I-}

  Reset(F);

{$I+}

  case IOResult of

    0: { All's well };

      try

        ...

      finally

        CloseFile(F);

      end;

    1..6, 15, 100..106: { Oops };

  end;

Incidentally, this error code table has some comments regarding possible causes of the various errors - this is mostly taken from Borland Pascal’s help system as Delphi 1’s help seems to be missing the run-time error messages.

Number
Meaning
Comments

2
File not found
The routines Reset, Append, Rename, and Erase report this error if the name assigned to the file variable does not specify an existing file.

3
Invalid path
The routines Reset, AppendFile, Rewrite, Rename, and Erase report this error if the name assigned to the file variable is invalid or specifies a non-existent subdirectory.

The routines ChDir, MkDir, and RmDir report this error if the path is invalid or if it specifies a non-existent subdirectory.

4
Too many files
The routines Reset, Rewrite, and Append report this error if the program has too many open files.

A program by default has access to 20 files including some system-reserved ones. A Delphi VCL app. has 255 (the maximum) available as the Delphi 1 VCL calls the SetHandleCount API for you.

5
File access denied
Reported by:
Occurs if:



Reset
FileMode allows writing and the name assigned to the file variable specifies a directory or a read-only file



Append
FileMode allows writing and the name assigned to the file variable specifies a directory or a read-only file



Rewrite
The directory is full, or the name assigned to the file variable specifies a directory or an existing read-only file



Rename
The name assigned to the file variable specifies a directory, or the new name specifies an existing file



Erase
The name assigned to the file variable specifies a directory or a read-only file



MkDir
A file with the same name exists in the parent directory, or there is no room in the parent directory, or the path specifies a device



RmDir
The directory is not empty, or the path does not specify a directory, or the path specifies the root directory



Read
The typed file is not open for reading



BlockRead
The untyped file is not open for reading



Write
The typed file is not open for writing



BlockWrite
The untyped file is not open for writing

6
Invalid file handle
This error is reported if an invalid file handle is passed to a DOS system call. This error should never occur; if it does, you know that the file variable has been corrupted.

15
Invalid drive number
GetDir and ChDir report this error if the drive number is invalid.

100
Disk read error
Read reports this error on a typed file if you attempt to read past the end of the file.

101
Disk write error
CloseFile, Write/Writeln, and Flush report this error if the disk becomes full.

102
File not assigned
Reset, Rewrite, Append, Rename, and Erase report this error if the file variable has not been assigned a name through a call to AssignFile.

103
File not open
These routines report this error if the file is not open: BlockRead, BlockWrite, CloseFile, Eof, FilePos, FileSize, Flush, Read, Seek, Write

104
File not opened for input
These routines report this error on a text file if the file is not open for input: Eof, Eoln, Read/ReadLn, SeekEof, SeekEoln

105
File not opened for output
This error occurs if you try to use standard I/O functions like ReadLn and WriteLn without the WinCrt unit used in your program.

106
Invalid numeric format
Read and ReadLn report this error if a numeric value read from a text file does not conform to the proper numeric format.

Error values from IOResult, run-time errors and an I/O exception’s ErrorCode property, common to Delphi 1 and 2

The example program FileH.Dpr on the CD shows an example of how to produce most of these errors. It also shows the various ways the errors may be reported. If compiled with I/O checking on (the default) you can get exceptions or run-time errors. If rebuilt with I/O checking disabled, IOResult is examined to find the problem which is then reported by a message box.

You might notice when running the program that the run-time errors don’t look much like run-time errors as earlier described - a run-time error handler has been installed to show you how to write one, should you wish to do so. As mentioned above, a routine has been written whose address can be assigned to ErrorProc. I won’t go into the details of how it works - it is supplied in the ErrHand.Pas unit and has some comments in. If you want to get back to normal run-time errors, find this assignment in FileHEtc.Pas:

ErrorProc := @ErrorHandler;

and change it back to:

ErrorProc := nil;

Suffice it to say that any software run-time error (i.e. not a GPF etc.) is passed to the routine which proceeds to raise a custom exception of type ERuntimeError. If the code and logic is not clear enough, let us know and we’ll expand on it another time
.

Win32 issues

Delphi 2 adds certain porting problems for I/O error handling. Although Win32 uses the same error codes as DOS, some error are reported more precisely and so different errors get generated. As an example, a call to ChDir(‘A:\’) where there is no floppy disk in drive A: will give an I/O error of 3 (path not found) in Delphi 1, but error 21 (ERROR_NOT_READY, device not ready) in Delphi 2. For a complete list of Win32 error codes load up the Delphi 2 help index and look for error codes and then choose Error Codes (Win32 Programmer's Reference), but be prepared for some missing text (many occurrences of ERROR are listed as ROR in the initial release).

File sharing

SetErrorMode

The first thing to say about file sharing is that Windows often displays quite unsightly error messages when programs try to do things that fail. If a 16-bit
 Delphi program tries to open a file that another process already has opened, Windows displays a dull, white Cancel/Retry message which must be cleared before your program displays its own error. To remove Windows from the equation, you use SetErrorMode. This API takes several constants in combination as a parameter and, when called, returns the old combination. It is usually a good idea to save this old value and restore it when you finish your file handling:

var

  OldMode: Word;

...

OldMode := SetErrorMode(sem_FailCriticalErrors or sem_NoOpenFileErrorBox);

AssignFile(F, ‘c:\delphi\readme.txt’);

try

  Reset(F);

except

  on EInOutError do

    {Handle problem};

end;

SetErrorMode(OldMode);

This code stops file sharing errors and file not found errors from being displayed. The other value that SetErrorMode accepts is sem_NoGPFaultErrorBox. In a Delphi program without hardware exception handling (e.g. a 16-bit WinCrt application that doesn’t anywhere use the SysUtils unit) this stops the system modal dialog that Windows shows when a GPF occurs.

FileMode

The default behaviour of Reset on a non-text file is to open it for reading and writing. This is dictated by the default value (2) of the FileMode variable. FileMode is passed to DOS by Reset to determine the way in which the file should be opened. The on-line help for FileMode lists three possible values:

0
open for reading only

1
open for writing only

2
open for both reading and writing

In fact there are several more values (for operating systems past DOS version 3.0), and even the documented ones have constants defined for them in SysUtils:

SysUtils constant
Value
Meaning

fmOpenRead
0
Open file read only

fmOpenWrite
1
Open file write only

fmOpenReadWrite
2
Open file for reading and writing

fmShareCompat
0
Open file in historic FCB-compatible way

fmShareExclusive
$10
Open file preventing others reading or writing

fmShareDenyWrite
$20
Open file preventing others writing

fmShareDenyRead
$30
Open file preventing others reading

fmShareDenyNone
$40
Open file allowing others to read and write

Usually one of the fmOpenxxxx constants is added to one of the fmSharexxxx constants. A common combination is to open a file for reading and writing, whilst preventing anyone else writing to the file. This could be achieved by doing this assignment before opening the file:

FileMode := fmOpenReadWrite + fmShareDenyWrite;

The other common expression allows your program and any other program to write to the file:

FileMode := fmOpenReadWrite + fmShareDenyNone;

Delphi 2 FileMode bug

Delphi 2.00 and 2.01 have a bug in Reset where FileMode gets ignored. If FileMode is set to a value greater than 2 (i.e. any of the sharing modes), it is considered to be 2 (i.e. normal read/write). This is fixed in a small update file available from the normal electronic distribution sites. SYSTEM.ZIP contains a replacement SYSTEM.DCU that can be placed in Delphi’s LIB directory to remove the problem. A simple recompile of the program is all that is necessary.

Record locking

When using database tables in a multi-user environment, it is well understood that you must lock a record before editing it to prevent anyone else simultaneously editing it and risking the integrity of the file. The same danger exists when using files that are not traditional database tables, using file variables and file handles. If you have a structured binary file that you use for random access via a file variable, then when writing data to a record it is wise to lock it if there is any danger of any other program attempting to write to the same file.

Locking a database table can be done with the TTable.LockTable method, whilst locking a file is generally done by setting FileMode to the value shown above or similar. Locking a table record in a multi-user environment is done by calling TTable.Edit, whilst locking a file record is rather more messy. There is no 16-bit Windows routine to help us out here - we need to call upon a DOS interrupt. Calling interrupts is completely non-portable, and in Delphi 1 the only high-level way to do it is with the essentially obsolete WinDos unit shipped in the \DELPHI\SOURCE\RTL70 directory. This unit implements a type TRegisters and two routines, Intr and MsDos. Instead of relying on old Borland Pascal code I will implement the routine with assembler statements.

The Win32 API does have a function for locking records called LockFile (and a corresponding UnlockFile). This allows you to lock any range of bytes in a file from any position (including past the end of the file, for when adding data). There is also a LockFileEx/UnlockFileEx pair, although these are not implemented in Windows 95, so we’ll forget about them. Anyway, the root routine for record locking will have a name and parameter set that match this API:

function LockFile(

  hFile: THandle; { handle of file to lock }

  { lock region offset, low and high words }

  dwFileOffsetLow,

  dwFileOffsetHigh,

  { length to lock low and high words }

  nNumBytesToLockLow,

  nNumBytesToLockHigh: Cardinal): Bool;

In Delphi 2, LockFile will allow you to lock any 64-bit sized range starting at any 64-bit offset in a file (Integer will be 32-bits), but currently it will do a 32-bit range from a 32-bit offset (as Integer is 16-bit). The implementation is found in the file NetLock.Pas and is:

asm

  mov ah, $5C

  mov al, 0

  mov bx, Handle

  mov cx, FileOffsetHigh

  mov dx, FileOffsetLow

  mov si, LockBytesHigh

  mov di, LockBytesLow

  int $21

  jnc @1

  xor ax, ax

  jmp @2

 @1:

  mov ax, 1

 @2:

end;

Because this function inconveniently forces the programmer to split values into two, there are other wrapper routines that take a file handle, or a file variable. If the latter is passed, you can specify the number of records to lock (the record size can be obtained by examining the file variable), or call a routine that locks just one record:

{ Handle based routine that un/locks given number of bytes }

function LockFileArea(Handle: Integer; FileOffset,(
  LockBytes: Longint; Lock: Boolean): Bool;

{ File variable based routine that }

{ un/locks a given number of records }

function LockFileVarArea(var FileVar; RecordNumber,(
  NumRecords: Longint; Lock: Boolean): Bool;

{ File variable based routine that un/locks one record }

function LockFileVar(var FileVar;(
  RecordNumber: Longint; Lock: Boolean): Bool;

The routine is demonstrated by two WinCrt programs, RECLOCK.DPR and RECLOCK2.DPR. Make sure both projects are compiled, and then run RECLOCK.EXE. This program makes a simple structured file with three strings in it. It then applies various locks to the file and each time launches RECLOCK2, which tries to access the file in various ways and reports its success or failure. The idea is to see the output in the second app, close it and then tell the first app that the second has closed by pressing a key, whereupon it will then do the next lock and run the second app again.

RECLOCK does this:

Opens file in read/write, deny write mode

Reopens file in read/write, deny none mode

Locks record 1

Unlocks record 1

Each time RECLOCK2 is launched it tries to

Open the file in read/write, deny none mode

If successful it tries to lock and then unlock record 1

When the file is opened in deny write mode, the file cannot be opened read/write by the second program. When opened in deny none mode, the file can be opened in read/write, deny none and the lock and unlock succeed. When record 1 is locked, the second application cannot lock record 1, nor unlock it.

File handles

A file handle is an integer number. Its meaning is not relevant for programmers other than to say that a value of -1 means that your file didn’t open because something went wrong. Error handling is thus fairly simple (although see the Error Handling section further on). We are advised usually not to refer to literal values like -1, but instead to use constants. When programming 16-bit Windows in C and C++ there is a constant defined called hFile_Error, which has a value of -1. Unfortunately this is one of the few items which has no equivalent definition in Delphi 1. Delphi 2 gives us both hFile_Error and the new Win32 constant for -1, which is Invalid_Handle_Value.

Apparently, a file handle is what Microsoft call a “magic cookie,” meaning you can use it to refer to a file without understanding its value. Historically in DOS (and this is also the case in Windows 3.1x) each program has an array of 20 bytes (though this number can be increased) called a Job File Table (JFT). A file handle is an index into your JFT. Each JFT entry was itself an index into DOS’s System File Table (SFT), whose size is dictated by the FILES= statement in CONFIG.SYS. Each SFT entry contains information about an open file including a file name, file position, file size and date and time stamps. A file handle is not a unique way of addressing a file - there can be several file handles all referring to the same file. If you cause three file handles to be allocated, it is your responsibility to ensure each one is closed even if they are all for the same file.

So in short, a file handle is a number. Remember that we can find a file variable’s file handle by using TFileRec(FileVar).Handle or TTextRec(TextVar).Handle. Conceptually you could consider operations using a file handle to be quite similar to operations on a file of byte, however file handle operations don’t raise exceptions. Additionally, you can write data blocks greater than 64k to a file via a file handle. Delphi 1 file variables can’t deal with data blocks over 64k in size. The file handle support routines in Delphi 1 and 2 are listed in the following tables.

SysUtils unit

FileClose
closes a file. Synonym for _lclose (Win16)/CloseHandle (Win32)

FileCreate
creates a new file and opens it, returning a file handle

FileGetDate
replaces GetFTime to get file modification time/date. Used with FileDateToDateTime

FileOpen
opens an existing file in a mode specified by file open mode constants

FileRead
reads data from a file. Synonym for _hread/ReadFile

FileSeek
moves file pointer position. Synonym for _llseek/SetFilePointer

FileSetDate
replaces SetFTime to set file modification time/date. Used with DateTimeToFileDate

FileWrite
writes data to a file. Synonym for _hwrite/WriteFile

Delphi file handle routines

WinProcs unit (Delphi 1 and Borland Pascal)

_hread
like _lread but handles blocks over 64kb. Not predefined in Delphi

_hwrite
like _lwrite but handles blocks over 64kb. Not predefined in Delphi

_lclose
closes a file

_lcreat
creates or opens a file

_llseek
repositions the file pointer

_lopen
opens a file

_lread
read data from a file

_lwrite
writes data to a file

OpenFile 
creates, opens, reopens or deletes a file

SetHandleCount
changes the number of file handles available to a task. The Delphi 1 Controls unit uses this API to set 255 file handles

Windows unit (Delphi 2) 

_hread
for compatibility with 16-bit Windows

_hwrite 
for compatibility with 16-bit Windows

_lclose 
for compatibility with 16-bit Windows

_lcreat 
for compatibility with 16-bit Windows

_llseek 
for compatibility with 16-bit Windows

_lopen 
for compatibility with 16-bit Windows

_lread 
for compatibility with 16-bit Windows

_lwrite 
for compatibility with 16-bit Windows

CloseHandle
close file

CreateFile
create/open/truncate a file

FlushFileBuffers
write files to disk

GetFileSize
find size of file

LockFile
lock area of a file

LockFileEx
not implemented in Windows 95

OpenFile 
for compatibility with 16-bit Windows

ReadFile
read data from a file

ReadFileEx
not implemented in Windows 95

SetEndOfFile
sets the current file position as the end of file

SetFilePointer 
repositions the file pointer

SetHandleCount 
for compatibility with 16-bit Windows

UnlockFile
unlock area of file

UnlockFileEx
not implemented in Windows 95

WriteFile 
writes data to a file

WriteFileEx
not implemented in Windows 95

Windows file handle routines

If you are still working with Delphi version 1.00 or 1.01 (your DELPHI.EXE time stamp won’t be 8:02) then your help page for FileOpen erroneously says that it is an internal routine. So does FileCreate’s page, but at least that describes the function. These mistakes were corrected in the 1.02 maintenance release, but for those without that version, here is the text (with a spelling correction):

FileOpen opens the specified file using specified access mode. The access mode is constructed by Or-ing one of the fmOpenXXX constants with one of the fmShareXXX constants. If the return value is positive, the function was successful and the value is the file handle for the opened file. If the return value is negative, an error occurred and the value is a negative DOS error code.
Error handling

Note that the return value for FileOpen (and indeed for some other routines) isn’t necessarily ‑1 for an error condition - it could be any of a range of negative numbers. This is in contrast to Delphi 2 where an error does yield -1 (or hFile_Error) and additional error information can be gleaned by using the GetLastError API. The reason for the difference is that Delphi 1 implements FileOpen, among other routines, by calling DOS interrupts. Calling interrupts is taboo in 32-bit, and so these routines are now implemented by calling appropriate Win32 APIs (i.e. CreateFile with appropriate parameters).

The likely DOS error values for the 16-bit FileOpen and FileCreate are as follows:

Value
Meaning

-2
File not found

-3
Path not found

-4
Too many open files

-5
Access denied

-12
Invalid access mode

File information

If you have been paying attention to what functionality was available for file variables, you may feel the 16-bit file handle support is a bit limited. There is no direct support for finding a file’s size, or the current file position. But fear not - we can still get hold of this information. The routine FileSeek (which is implemented by a call to _llseek in Win16 or SetFilePointer in Win32) can move to any place in your file, and then return that file position. We specify where to move to by giving it an offset (a number of bytes) and a symbol indicating an origin to seek from which can mean from the beginning of the file (0), from the current position (1) or from the end of the file (2). The symbols have different constant names, depending which function you are calling:

Origin
Value
FileSeek constant
_llseek constant
SetFilePointer constant

Beginning of file
0
soFromBeginning
Seek_Set
File_Begin

Current position
1
soFromCurrent
Seek_Cur
File_Current

End of file
2
soFromEnd
Seek_End
File_End

To find our current position, we can seek zero bytes from the current position, and we will be told how far through the file we are. To find the file size, we need to do a similar thing, but save the current position, then seek zero bytes from the end of the file - recording the position which will be the file size. To get back to where we were we can seek from the beginning of the file, specifying our saved position as the number of bytes.

function GetFileSize(Handle: Integer): Longint;

var

  FileSize: Longint;

begin

  Result := FileSeek(Handle, 0, soFromCurrent);

  if Result > -1 then

  begin

    FileSize := FileSeek(Handle, 0, soFromEnd);

    FileSeek(Handle, Result, soFromBeginning);

    Result := FileSize;

  end;

end;

Words of warning

When dealing with files, the data types used for storage need to be thought about if portability of data files between 16-bit and 32-bit Delphi programs is to be maintained. If integral values are being stored, remember to use Smallint or Word, rather than Integer or Cardinal for signed and unsigned 16-bit numbers respectively. Delphi 2, like Delphi 1, interprets Smallint and Word as 16-bit values, whereas Integer and Cardinal become 32-bit values in Delphi 2.

On another vein, Delphi 2 records get laid out in memory differently by default. Instead of each field immediately following the previous one, Delphi 2 ensures each one starts at a suitable boundary (dependent on the field size) for efficient access. This means there may be spare bytes in your records which will cause the record size to increase, and break programs that read data from Delphi 1 days. To prevent problems, precede to word record with the word packed (this used to be ineffective, but is now significant) or the $Align compiler directive.

Here is an example:

type

TUnsafeRecord = record

  Ch: Char;

  L: Longint;

  B: Boolean;

end; { 12 bytes in Delphi 2, 6 bytes in Delphi 1 }

TSafeRecord = packed record

  Ch: Char;

  L: Longint;

  B: Boolean;

end; { 6 bytes in Delphi 1 & 2 }

The safe version could also be written like this, or using {$A-} and {$A+} for {$Align Off} and {$Align On} respectively.

{$Align Off}

TSafeRecord = record

  Ch: Char;

  L: Longint;

  B: Boolean;

end; { 6 bytes in Delphi 1 & 2 }

{$Align On}

The problem with using these compiler directives is that the $Align directives are new in Delphi 2, and so would be rejected by Delphi 1. Also the $A directives were global in Delphi 1, not local as in Delphi 2, and so couldn’t be used as above in code for both products.

Also string data needs to be considered carefully. If you have 16-bit programs writing data out to non-text files which include strings, ensure you either disable huge string support for the areas of your Delphi 2 program that deal with the file I/O, or alternatively explicitly declare your strings as short strings (e.g. S: String[255] - the length limit in square brackets makes a Delphi 1 compatible short string).

Another gotcha with strings in Delphi 2 comes up with FileWrite. The second parameter of FileWrite is an untyped variable - you pass some data, and it takes the address of that data and passes it through. In Delphi 1 you can pass a string in, maybe something like this:

var

  Handle: Integer;

  S: String;

...

  Handle := FileCreate('c:\deleteme.dat');

  FileClose(Handle);

...

  Handle := FileOpen('c:\deleteme.dat',(
    fmOpenReadWrite or fmShareDenyNone);

  S := Edit1.Text;

  FileWrite(Handle, S, SizeOf(S));

...

  FileSeek(Handle, 0, soFromBeginning);

  FileRead(Handle, S, SizeOf(S));

  FileClose(Handle);

  Caption := S;

This writes a whole string variable out - a 256 byte block of data. Because of the use of a var parameter, things turn pear-shaped in Delphi 2. The new huge strings are implemented via an implicit pointer. A string variable is really a pointer to the string data. If you try and pass a huge string to FileWrite, it passes the address of the pointer and all you get in the file is the value of that pointer plus a load of garbage. Instead you need to pass S[1], so the address of the first character would be passed.

If a Delphi 1 application wrote a string out as above, a Delphi 2 huge string program would need to be fiddly to match its operation and maintain the file structure and layout. Strings are managed by dynamic allocation of memory, which is increased when a string is written to using normal string operations, but not using a memory write operation as FileRead does. Bearing this in mind, we could use:

var

  Handle: Integer;

  S: String;

  Len: Byte;

...

  Handle := FileCreate('c:\deleteme.dat');

  FileClose(Handle);

...

  Handle := FileOpen('c:\deleteme.dat',(
    fmOpenReadWrite or fmShareDenyNone);

  S := Edit1.Text;

  Len := Length(S);

  FileWrite(Handle, Len, SizeOf(Byte));

  SetLength(S, 255);

  FileWrite(Handle, S[1], 255);

  SetLength(S, Len);

...

  FileSeek(Handle, 0, soFromBeginning);

  FileRead(Handle, Len, SizeOf(Byte));

  SetLength(S, 255);

  FileRead(Handle, S[1], 255);

  SetLength(S, Len);

  FileClose(Handle);

  Caption := S;

The first SetLength in the writing section makes sure there are definitely 255 valid bytes which FileWrite can write to the file. If there were only, say, 5, then we would risk an access violation. The second one restores the string to its old length. The first SetLength in the reading section causes enough memory to be available to write the 255 characters to, and then the second one ensures that the string thinks of itself with the correct length. However, the string storage here is inefficient - 256 bytes for each string. Let’s change the Delphi 1 writing and reading code to this:

  FileWrite(Handle, S, Length(S) + 1);

...

  FileRead(Handle, S[0], 1);

  FileRead(Handle, S[1], Length(S));

The Delphi 2 reading code now turns into:

  Len := Length(S);

  FileWrite(Handle, Len, 1);

  FileWrite(Handle, S[1], Length(S));

...

  FileRead(Handle, Len, 1);

  SetLength(S, Len);

  FileRead(Handle, S[1], Len);

32-Bit considerations - Long File Names and UNCs

If you are considering the move to Delphi 2, you will be wondering about long file names, Universal Naming Conventions (UNCs) and Unicode. Long file names and UNCs are transparently supported on operating systems that support them. So a Delphi 2 app running on Windows 95 could make a file on a local hard disk called HelloWorldThisIsALongName, but couldn’t make the same file on a network server running an operating system that doesn’t support them.

Delphi 2 has a couple of new routines that specifically support UNCs: ExpandUNCFileName and ExtractFileDrive. It also adds ExtractFileDir which produces the same results as ExtractFilePath, but ensures there no trailing backslash.

A problem with all these Delphi routines that are introduced in this section is that none of them are Unicode-aware. All the routines take string parameters, and Delphi 1 and 2 only supports non-Unicode Pascal strings (which are synonymous to AnsiStrings, where each character is an AnsiChar - one byte). The Unicode WideString type (each character is a WideChar - two bytes) is expected to be implemented in a later version.

The occurrence of a currently unused symbol tkLWString in the RTTI defining TypInfo unit (see The TypInfo unit and its data types part of the Run-Time Type Information (RTTI) section) enforces this prediction of what is currently the future. tkChar represents a Char or AnsiChar, tkWChar represents a WideChar, tkString represents a short string, tkLString represents a long string. tkLWString will be used for a long wide character string, i.e. a Unicode string.

In the interim period you will be forced to use the appropriate Windows API to deal with Unicode filenames, i.e. CreateFileW. Remember, though, that Windows 95 has hardly any useful Unicode support - all the Unicode functions are there, they simply return with a failure error code.

More file handling routines

We will not be looking at the Windows file handling APIs as they tend to vary between Windows version, however they have been listed in the tables so you can look up their details in the Delphi online help. 

To give you more food for thought, the tables below list a plethora of file-oriented bits and pieces, including miscellaneous routines, directory and disk routines and also constants and data structures.

System unit (Delphi and Borland Pascal)

IOResult
returns value of last I/O error if I/O checking is disabled

SysUtils unit

ChangeFileExt
given a file name, this returns a string with the extension changed

DateTimeToFileDate
replaces PackTime to turn Delphi date/time into DOS date/time. Used with FileSetDate

DeleteFile
deletes a file

ExpandFileName
returns absolute file specification. Replaces FileExpand, FExpand

ExtractFileExt 
given a full file specification, returns the extension. Replaces FileSplit, FSplit

ExtractFileName 
given a full file specification, returns the name including extension. Replaces FileSplit, FSplit

ExtractFilePath
given a full file specification, returns the path. Replaces FileSplit, FSplit

FileAge
used with FileDateToDateTime returns age of file

FileDateToDateTime
Replaces UnpackTime to turn DOS date/time into Delphi date/time for FileAge, FileGetDate, TSearchRec

FileExists
returns True if file exists

FileGetAttr
replaces GetFAttr for finding file attributes

FileSearch
locates a file on a given path

FileSetAttr
replaces SetFAttr for setting file attributes

FindClose
terminates a FindFirst, FindNext sequence

FindFirst
finds first occurrence of a file specification (can handle wildcards)

FindNext
finds next occurrence of a file specification (can handle wildcards)

RenameFile
renames a file

SysUtils unit (Delphi 2)

ExtractFileDir
much the same as ExtractFilePath but leaves no trailing \ character

ExtractFileDrive
given a full file specification, returns the drive portion. Caters for drive paths and UNC paths

ExpandUNCFileName
same as ExpandFileName but uses UNC format to the path if applicable

WinDos unit (Delphi 1 and Borland Pascal)

FileExpand
returns absolute file specification. Delphi introduces ExpandFileName

FileSearch
locates a file on a given path

FileSplit
splits a file specification into path, directory, name and extension. Delphi introduces ExtractFileExt, ExtractFileName and ExtractFilePath

FindFirst
finds first occurrence of a file specification (can handle wildcards)

FindNext
finds next occurrence of a file specification (can handle wildcards)

PackTime
became DateTimeToFileDate. Used with SetFTime to turn DOS date/time to DateTime or TDateTime

UnpackTime
became FileDateToDateTime. Used with GetFTime, TSearchRec, SearchRec to turn DateTime or TDateTime into DOS date/time

Windows unit (Delphi 2)

DeleteFile
Deletes a file

CopyFile
Copies an existing file to a new file

FindClose
Terminates a FindFirstFile, FindNextFile sequence. Note the same name as the System unit procedure

FindFirstFile
finds first occurrence of a file specification (can handle wildcards)

FindNextFile
finds next occurrence of a file specification (can handle wildcards)

GetBinaryType
identifies application type (e.g. Win16, Win32, DOS, OS/2)

GetFileAttributes
gets attributes of a file or directory

GetFileTime
gets file time stamp

GetFullPathName
returns full path and file name of a file (even an 8.3 file name)

GetShortPathName
returns the short path form of the specified input path

MoveFile
renames a file or directory

MoveFileEx 
not implemented in Windows 95

SearchPath
searches for a specified file

SetFileAttributes
sets file attributes

SetFileTime
sets file timestamp

Miscellaneous routines

System unit (Delphi and Borland Pascal)

ChDir
changes current directory. Replaces CreateDir

GetDir
gets current directory. Replaces GetCurDir

MkDir
makes new directory. Replaces CreateDir

RmDir 
removes a directory. Replaces RemoveDir

SysUtils unit

DiskFree
returns free disk space

DiskSize
returns disk size

SysUtils unit (Delphi 2)

CreateDir
same as MkDir but doesn’t cause an exception on error

GetCurrentDir
like GetDir but only for current drive

RemoveDrive
same as RmDir but doesn’t cause an exception on error

SetCurrentDir
same as ChDir but doesn’t cause an exception on error

WinProcs unit (Delphi 1 and Borland Pascal)

GetDriveType
determines if a drive is removable, fixed or remote

GetSystemDirectory
retrieves Windows system directory path

GetTempDrive
returns a drive letter where temporary files may be stored

GetTempFileName
creates a temporary file

GetWindowsDirectory
retrieves Windows directory path

WinDos unit (Delphi 1 and Borland Pascal)

DiskFree
returns free disk space

DiskSize
returns disk size

CreateDir
changes current directory. Delphi introduces MkDir

GetCurDir
gets current directory. Delphi introduces GetDir

RemoveDir
makes new directory. Delphi introduces RmDir

SetCurDir
removes a directory. Delphi introduces ChDir

Windows unit (Delphi 2)

CreateDirectory
makes a new directory

CreateDirectoryEx
makes a new directory with the attributes of another directory

GetCurrentDirectory
gets current directory

GetDiskFreeSpace
gives information about free space on disk

GetDriveType
identifies drive type

GetLogicalDrives
identifies currently available drives

GetLogicalDriveStrings
returns names of currently available drives

GetSystemDirectory
retrieves Windows system directory path

GetTempFileName
creates a temporary file

GetTempPath
returns a path where temporary files may be stored

GetVolumeInformation
gives various information about a file system and volume

GetWindowsDirectory 
retrieves Windows directory path

RemoveDirectory
removes a directory

SetCurrentDirectory
changes current directory

SetVolumeLabel
changes file system volume label

Directory/disk routines

System unit (Delphi and Borland Pascal)

File
untyped file type

File of xxxx
typed file type, e.g. File of Double

FileMode
affects how Reset opens a file

Text
text file type

TextFile
Delphi substitute for Text to avoid scoping problems

SysUtils unit

faxxxx
used by FileSetAttr, FileGetAttr, TSearchRec, e.g. faReadOnly

fmxxxx
file open mode constants, used by FileMode variable, e.g. fmOpenReadWrite

fmxxxx
file mode constants, used by TTextRec and TFileRec, e.g. fmClosed 

TFileName
generic file name type

TFileRec
non-text file internal representation

TSearchRec
search record used by FindFirst, FindNext

TTextRec
text file internal representation

Classes unit

soFromxxxx
seek origin constants for FileSeek and TStream.Seek e.g. soFromEnd

WinTypes unit (Delphi 1)

Seek_xxxx
seek origin constants for _llseek, e.g. Seek_End

Windows unit (Delphi 2)

File_xxxx
seek origin constants for SetFilePointer/_llseek, e.g. File_End

WinDos unit (Delphi 1 and Borland Pascal)

DosError
where DOS file errors are reported

faxxxx
file attribute constants, used by GetFAttr, SetFAttr, TSearchRec, e.g. faReadOnly

fcxxxx
file component constants, used by FileSplit, e.g. fcDirectory

fmxxxx
file open mode constants and file mode constants, used by TTextRec and TFileRec, e.g. fmClosed

fsxxxx
file name component string lengths, e.g. fsPathName, fsExtension

TDateTime
used by GetFTime, SetFTime, PackTime, UnpackTime

TFileRec
non-text file internal representation

TSearchRec 
search record used by FindFirst, FindNext

TTextRec
text file internal representation

File types, variables and constants

FileSize with no file variable or handle

To find out a file’s size, you can use FileSize which takes an open typed or untyped file variable. This returns the number of records in the typed file or bytes in the untyped file. Alternatively, if you have a file handle, you could use a few calls to FileSeek (as done in the GetFileSize routine shown in the earlier section File information). But what if you just want to find the size of an arbitrary file, without declaring a file variable or file handle? Try this routine.

function SizeOfFile(const FileName: String): Longint;

var

  SearchRec: TSearchRec;

begin

  Result := -1;

  if FindFirst(FileName, faArchive, SearchRec) = 0 then

    Result := SearchRec.Size;

  FindClose(SearchRec);

end;

Re-implementing FileExists

Because of some issues that you can come up against with the FileExists routine in SysUtils (implemented by checking a call to FileAge against the supplied file name), here is an alternative that many people use.

function FileExists(const FileName: String): Boolean;

var

  F: File;

begin

  AssignFile(F, FileName);

{$I-}

  Reset(F);

{$I+}

  Result := IOResult = 0;

end;

The trouble with it is that it fails on files that are read-only. Here are three more candidates using Reset, FileGetAttr and FindFirst.

function FileExists1(const FileName: String): Boolean;

var

  OldMode: Byte;

  F: File;

begin

  OldMode := FileMode;

  FileMode := fmOpenRead or fmShareDenyNone;

  AssignFile(F, FileName);

{$I-}

  Reset(F);

{$I+}

  FileMode := OldMode;

  Result := IOResult = 0;

end;

function FileExists2(const FileName: String): Boolean;

var

  FAttr: Integer;

begin

  FAttr := FileGetAttr(FileName);

  Result := (FAttr >= 0) and

            (FAttr and (faDirectory or faVolumeID) = 0);

end;

function FileExists3(const FileName: String): Boolean;

var

  SearchRec: TSearchRec;

begin

  if FindFirst(FileName, faAnyFile, SearchRec) = 0 then

    Result := SearchRec.Name = FileName;

  FindClose(SearchRec);

end;

The RTL FileExists can take wildcards, but the versions listed here do not. You can change FileExists3 to understand wildcards like this.

function FileExists3(const FileName: String): Boolean;

var

  SearchRec: TSearchRec;

begin

  Result := FindFirst(FileName, faAnyFile, SearchRec) = 0;

  FindClose(SearchRec);

end;

Streaming objects

Having looked at file variables and file handles, we will now turn out attention to streams, with a slight digression on our way through. These possibly unfamiliar items are used by Delphi developers day in and day out under the guise of form files, but more of that later.

What is a stream?

A stream is intended to be a generic interface to any storage medium. The storage medium could be, amongst other possibilities, a block of memory, a disk file, a database BLOb field or a Windows resource file. In Delphi there are several stream types, all unsurprisingly represented by objects. They are all derived from the abstract class (i.e. useless as a class for instantiating objects, but useful as a basis for deriving more specific classes) TStream.

Basic stream capabilities

Any Delphi stream object has at least two properties. The current position within the stream is given by the Position property, and the size of the stream is given by Size. To move to a different position in the stream, there is a method called Seek (rather similar to FileSeek, as used with file handles). This takes an offset to move by, and another parameter indicating where to move from. This can be one of the following symbols:

Symbol
Implication for the Offset Parameter

soFromBeginning
relative to the beginning of the stream

soFromCurrent
relative to the current stream position

soFromEnd
relative to the end of the stream

If Seek is successful, it returns the new position in the stream.

To write to the stream, you call the Write method which takes an untyped const parameter that acts as a buffer of information, and a number of bytes to write. It will return the number of bytes written. The implication of the parameter being untyped is that any data item can be passed. The implementation of a const/var parameter where a variable is passed (a pass by reference parameter) is that the address of the actual argument is passed (transparently to the programmer). If there is no type supplied in the parameter list of the subroutine definition, then any variable will be accepted and its address gets passed along to the routine. The Read method takes an untyped var parameter and a byte count, and returns the number of bytes read.

In addition to this basic set of functionality, a stream object also has a CopyFrom method which knows how to read data from another stream, and also methods to write and read components to and from itself. The WriteComponent and ReadComponent methods are complemented by WriteComponentRes and ReadComponentRes that write and read a standard Windows resource-file header before the component respectively. These can be used to manufacture custom resource files filled with components that can be linked into your executable.

Types of streams

Delphi 1.0x offers a number of specific stream types, and Delphi 2 add another one to the list. A new stream object to implement shared memory across Win32 processes is implemented in the Win32 chapter section Shared Memory Between Processes And Threads.

Stream class
Comments

THandleStream
Takes an existing open file handle and allows you to access the file as a stream

TFileStream
Takes a filename and a file access mode and treats the file as a stream. Based on a THandleStream

TBlobStream
Takes a BLOb field and a BLOb access mode (bmRead, bmWrite or bmReadWrite). Used to manipulate BLOb field data

TMemoryStream
Treats a block of memory as a stream

TResourceStream
New in Delphi 2. Takes an executable’s instance handle, a resource name/number and a resource type and lets you access a Windows resource as a stream

A THandleStream (and consequently a TFileStream) have a read-only Handle property to surface the file handle. When creating a TFileStream object, you pass the filename to the constructor, and also a file access mode. These modes were discussed in the last issue, but there is an additional one available. If you wish your file stream to represent a newly created file, then you can use the fmCreate mode, and it will endeavour to honour that request.

A TMemoryStream has a Memory property which returns a pointer to the beginning of the memory block. To start with it has no memory allocated and so Memory will be nil. You can make a memory stream as large as you like using the SetSize method, however be warned that this disposes of any currently allocated memory through a call to its Clear method. Normally when writing to a memory stream, if it finds it hasn’t got enough memory, it will allocate as much as is needed, rounding up to the nearest 8 KB. It is good practice to call SetSize straight after constructing a memory stream, passing a value that is as large as you need in the short term future, thus preventing lots of reallocations.

A TResourceStream also has a Memory, although this isn’t used much. This object is designed for reading components from custom resources in your executable. We’ll have a look at this later. To see a resource stream in operation (though not reading components) and also a memory stream, refer to the Bitmaps From Resources: 256+ colour bitmaps section in the VCL chapter. Note that the class in that section relies on the fact that a TBitmap object can set itself up from data on a stream, by using its LoadFromStream method. Many objects have a LoadFromStream and a corresponding SaveToStream method to allow storage or retrieval of their data to/from a stream. Here is a list of those that do:

TGraphic and its descendants TIcon, TMetafile and TBitmap
TStrings and its descendent TStringList (which principally means that TListboxes, TComboBoxes, TQuerys and many other objects have, since they all have TStrings properties)
TMemoryStream
TBlobField and its descendants TGraphicField and TMemoField
TOleContainer
TOutline
TTreeView (Delphi 2 only)

Copying files with streams

We can copy files around quite simply using a pair of streams. Apart from a pair of try..finally blocks to protect the TFileStream objects against exceptions, the code is one call to CopyFrom. Note that the code takes advantage of the fact that CopyFrom takes a zero as the byte count parameter to mean copy the whole stream. The project COPYEg3.DPR shows this in use, where the function is implemented in CopyU3.Pas.

procedure FileCopy(const InFileName, OutFileName: String);

var

  InStream, OutStream: TFileStream;

begin

  InStream := TFileStream.Create(InFileName,(
    fmOpenRead + fmShareDenyWrite);

  try

    OutStream := TFileStream.Create(OutFileName, fmCreate);

    try

      OutStream.CopyFrom(InStream, 0);

      FileSetDate(OutStream.Handle, FileGetDate(InStream.Handle));

    finally

      OutStream.Free;

    end;

  finally

    InStream.Free;

  end;

end;

This becomes even shorter when using a TMemoryStream. The following code comes from COPYEG4.DPR. Since a memory stream doesn’t surface any file handle, this code elects to not bother setting the file date.

procedure FileCopy(const InFileName, OutFileName: String);

begin

  with TMemoryStream.Create do

    try

      LoadFromFile(InFileName);

      SaveToFile(OutFileName);

      { Doesn't set date in this case, }

      { as there is no handy file handle }

    finally

      Free;

    end;

end;

Streaming data

Reading from and writing to a stream is reasonably straightforward for normal data types, and matches up quite well with file handle operations. For example, to write a variable called I to a stream you would use:

MyStream.Write(I, SizeOf(I));

and to read it back in would require:

MyStream.Read(I, SizeOf(I));

If you have delved into pointers and, for example, you have a pointer to a Longint called P, you could use:

MyStream.Write(P^, SizeOf(Longint));

...

MyStream.Read(P^, SizeOf(Longint));

Making objects streamable

If you are interested in streaming objects then things get a little more involved. Objects do not inherently know how to stream themselves. Objects that do know how are termed persistent objects - their data can potentially live on after a program has finished. There is a type called TPersistent in the VCL that is intended to act as a basis for all persistent objects, all objects that can potentially be streamed. Unfortunately, all the VCL-supplied mechanics for reading and writing persistent objects operate solely on objects based on a descendent of TPersistent called TComponent. So components can be streamed in a standard and well-supported way (discussed later), but other objects can’t. Let’s look into how we can get any arbitrary object into and out of a stream.

Object streaming project version 1

What we need to do rather depends on what we want to achieve. If we want a stream filled with the data of many objects of the same class, then we can add our own LoadFromStream and SaveToStream methods directly to that class. Here’s a simple example to demonstrate this which makes use of a TList populated with TPointData objects, where TPointData holds an X and Y co-ordinate. To make the class have a point (pun not intended) it has a method to swap the X and Y co-ordinates and also a choice of constructors. The normal constructor, Create, is inherited from type TObject, and if it is called, X and Y will be left with their default values of 0 (all object data fields are initialised with zeros). The alternative constructor CreateXY takes two parameters and sets X and Y with those values.

Our TPointData class will not be derived from type TPersistent since we won’t be using the VCL-supplied streaming mechanism (we don’t have TComponent as an ancestor, so we can’t use it). This example is supplied in the project Strm1.Dpr, and the interesting code is in Strm1U.Pas. The object’s definition and the stream access methods are listed here.

TPointData = class

  public

    X, Y: Word;

    constructor CreateXY(AX, AY: Word);

    procedure SwapXY;

    procedure LoadFromStream(Stream: TStream);(
      virtual;

    procedure SaveToStream(Stream: TStream); virtual;

  end;

...

procedure TPointData.LoadFromStream(Stream: TStream);

begin

  Stream.Read(X, SizeOf(X));

  Stream.Read(Y, SizeOf(Y));

end;

procedure TPointData.SaveToStream(Stream: TStream);

begin

  Stream.Write(X, SizeOf(X));

  Stream.Write(Y, SizeOf(Y));

end;

The program that uses this class is shown below. The Generate button makes a random number of TPointData objects and stores them in a TList object called PointList. The Save button iterates through the list and calls the SaveToStream method for each PointData object, and then empties the list. The Load button reads through the stream, creating an object and calling its LoadFromStream method until the end of the stream is reached. Lastly, the Swap button will iterate through the list and call the objects’ SwapXY methods. Whenever points are made or altered, they are drawn in a paint box on the form.



#####STREAM1.BMP

The useful code is as follows.

....

procedure TForm1.ClearPoints;

begin

  while PointList.Count > 0 do

  begin

    TPointData(PointList[0]).Free;

    PointList.Delete(0);

  end;

end;

procedure TForm1.PaintBox1Paint(Sender: TObject);

begin

  for Loop := 0 to PointList.Count - 1 do

  begin

    Pt := TPointData(PointList.Items[Loop]);

    if Loop = 0 then

      PaintBox1.Canvas.MoveTo(Pt.X, Pt.Y)

    else

      PaintBox1.Canvas.LineTo(Pt.X, Pt.Y)

  end;

end;

procedure TForm1.MakeBtnClick(Sender: TObject);

begin

  ClearPoints;

  for Loop := 1 to 20 do

  begin

    Pt := TPointData.CreateXY(Random((
      PaintBox1.Width), Random(PaintBox1.Height));

    PointList.Add(Pt);

    PaintBox1.Invalidate;

  end;

end;

procedure TForm1.SaveBtnClick(Sender: TObject);

var

  Stream: TFileStream;

begin

  Stream := TFileStream.Create(DataFile, fmCreate);

  try

    for Loop := 0 to PointList.Count - 1 do

    begin

      Pt := TPointData(PointList.Items[Loop]);

      Pt.SaveToStream(Stream);

    end;

  finally

    Stream.Free;

  end;

  ClearPoints;

  PaintBox1.Invalidate;

end;

procedure TForm1.LoadBtnClick(Sender: TObject);

var

  Stream: TFileStream;

begin

  ClearPoints;

  Stream := TFileStream.Create(DataFile,(
    fmOpenRead or fmShareDenyWrite);

  try

    while Stream.Position <> Stream.Size do

    begin

      Pt := TPointData.Create;

      Pt.LoadFromStream(Stream);

      PointList.Add(Pt);

    end;

  finally

    Stream.Free;

  end;

  PaintBox1.Invalidate;

end;

.....

Adding a SaveToStream and LoadFromStream method to an object allows its data to be streamed, and this approach is just as applicable if you have a set number of custom objects of different types that need to be streamed in a particular order.

However, this is not really all that object streaming can be. So far, when reading from the stream we need to know beforehand what object is next in the stream, so that we can construct one, and then call one of its methods to get it to read its own data. Object streaming can be much more flexible. Delphi does support this extra flexibility but only for classes derived from type TComponent.

Design-time stream requirements

One of the main uses of an object stream is to remove repetitive functionality from a program. If your program uses a set of objects each time it runs, and when it starts it initialises those objects, then your program is wasting a certain amount of time each run. Much better would be to use another program to set up the objects and write them to a stream, and then your program can just read the stream in. This is exactly what Delphi does. As described in the Run-Time Type Information section, it makes an object stream (a form file) to represent each of your forms at design time (when your program is not running). Then, rather than you having to set up all your components each time you run your program, the streams are read in as needed when forms are created and all the objects on them are automatically created, with all their properties set up.

Because components and forms are derived from type TPersistent Delphi is able to work them into form files by using their RTTI. In case you didn’t read the RTTI stuff, here is a brief recap:

1
Delphi writes into a form properties that appear on the Object Inspector

2
Properties appear on the Object Inspector only if they are declared in the published section of an object

3
An object can only have a published section if the compiler has been told to generate RTTI for it, or one of its ancestors

4
The $M+ or $TypeInfo On compiler directives tell the compiler to generate RTTI

5
The default state of the RTTI generation is off, to avoid masses of the stuff being made (it gets stored in your EXE). It appears that this is not actually the case in Delphi 2, but we’ll try and forget that for the purposes of this overview.

6
TObject is compiled in a $M- state (if you make a class inherited from TObject and put a published section in, you will get Error 200: PUBLISHED not allowed in this class from Delphi 1, although Delphi 2 appears to erroneously let it go through)

7
TPersistent is compiled in the $M+ state

This means that anything derived from TPersistent can have a published section, and type information will be available for entries in that section. The Object Inspector makes use of this information to show the properties and allow their values to be changed. The form designer makes use of the information to save the properties to a form file and to read them back in again. This functionality is available to you as a programmer (see below).

Streaming components

Delphi ForM Files

Delphi uses RTTI to produce and maintain DFM files at design time. The DFM file is a Windows resource file with a custom resource in it. It is linked into your program by the $R compiler directive at the top of your form unit’s implementation section. {$R *.DFM} doesn’t mean link in all DFM files, but refers to the DFM file with the same name as this unit. The custom resource in the form file is an object stream - the form object’s non-default property values and all the non-default properties of all the objects on the form.

You can manufacture a file exactly the same as a DFM file at run-time using WriteComponentResFile from the Classes unit, as shown below. If you place this code in a form’s OnClick handler and click on the form at run-time, you get an FRM file that will exactly match your Delphi DFM file. Running CONVERT.EXE over both of them will give the same text file version of the form.

{ At run-time, a Delphi 1 form’s name }

{ property is blank. Delphi 2 remedies this }

if Name <> '' then

  Name := 'Form1';

{ Visible defaults to False at design time }

Hide;

try

  { ActiveControl was blank at design time }

  ActiveControl := nil;

  { Write out a form file - UNIT1.FRM }

  { should end up the same as UNIT1.DFM }

  WriteComponentResFile('UNIT1.FRM', Form1);

finally

  { Unhide the form }

  Show;

end;

WriteComponentResFile simply constructs a temporary TFileStream and calls its WriteComponentRes method, using the component’s class name (as returned by the ClassName method) as the resource name, and the component as the resource. WriteComponentRes in turn writes a Windows resource header and then calls the stream’s WriteComponent method which again delegates the hard work to someone else. It constructs a TWriter object and calls its WriteRootComponent method. When writing a form file, the form is the root component as it owns all components which also need to be written to the stream - i.e. all the components on the form.

When a form is created at run-time in a Delphi application, the form resource is read in and the property values are read in. You can do this manually if you feel like it using WriteComponentResFile’s counterpart ReadComponentResFile. Assuming you have the relevant form class in your program somewhere, this code will do the trick:

var

  AForm: TForm;

…

begin

  if FileExists('Other.Dfm') then

  begin

    AForm := TOtherForm.CreateNew(Application);

    ReadComponentResFile('Other.Dfm', AForm);

    AForm.ShowModal;

    AForm.Free

  end

The code that usually does this lives in the VCL. When you open a form file in the Delphi editor, or use the DOS CONVERT.EXE tool, you can translate between a form file and a text file. This facility is also in the VCL. The important routines are ObjectTextToBinary, ObjectBinaryToText, ObjectResourceToText and ObjectTextToResource all of which take an input stream and an output stream. On the CD is a project called ReadFrm.Dpr which shows how to generate a text file from a form file (ObjectResourceToText) and also from a form resource (ObjectBinaryToText).



If compiled in Delphi 1, the code uses a THandleStream. If in Delphi 2, it uses the new TResourceStream. The two routines from the form unit ReadFrmU are shown here.

procedure TFormReader.Button1Click(Sender: TObject);

var

  InStream,

  OutStream: TFileStream;

begin

  InStream := TFileStream.Create(FormFile, fmOpenRead);

  try

    OutStream := TFileStream.Create(TextFile, fmCreate);

    try

      { Translate a resource file to a text file }

      ObjectResourceToText(InStream, OutStream);

    finally

      OutStream.Free;

    end;

  finally

    InStream.Free;

  end;

  FormDescription.Lines.LoadFromFile(TextFile);

end;

procedure TFormReader.Button2Click(Sender: TObject);

var

  InStream: {$ifdef VER80}THandleStream{$else}

                          TResourceStream{$endif};

  OutStream: TFileStream;

begin

{$ifdef VER80}

  InStream := THandleStream.Create(AccessResource((
    HInstance, FindResource(HInstance, 'TFormReader', rt_RCData)));

{$else}

  InStream := TResourceStream.Create(HInstance,(
   'TFormReader', rt_RCData);

{$endif}

  try

{$ifdef VER80}

    if InStream.Handle = 0 then

      raise EResNotFound.CreateResFmt(SResNotFound, [ClassName]);

{$endif}

    try

      OutStream := TFileStream.Create(TextFile, fmCreate);

      try

      { Translate exe-based resource to a text file }

        ObjectBinaryToText(InStream, OutStream);

      finally

        OutStream.Free;

      end;

    finally

      InStream.Free;

    end;

  finally

    FileClose(Handle);

  end;

  FormDescription.Lines.LoadFromFile(TextFile);

end;

Streaming custom components (or object streaming project version 2)

We will now return to the STRM1.DPR project from earlier, which defined basic objects and streamed them using a custom SaveToStream method and LoadFromStream method. The idea of using components is that we can somehow improve the situation. The problem before was that we needed to know what objects were in the stream, and in what order, to get them back out again. Also we were responsible for creating all the objects whose data was read back in. A form in a DFM file does not have this requirement - all components on the form are automatically manufactured when the form is read in.

As the first stage along the road of improvement, we will simply redeclare the TPointData structure as a component. Since components have the capability of being owned, we ensure that the constructor takes an owner component as a parameter. This means we don’t need to delete any outstanding TPointData objects at the end of the program run - someone else (the owner) will take care of that.

  TPointData = class(TComponent)

  private

    FX, FY: Word;

  public

    constructor CreateXY(AOwner: TComponent; AX, AY: Word);

    procedure SwapXY;

  published

    property X: Word read FX write FX default 0;

    property Y: Word read FY write FY default 0;

  end;

...

constructor TPointData.CreateXY(AOwner: TComponent; AX, AY: Word);

begin

  inherited Create(AOwner);

  FX := AX;

  FY := AY;

end;

procedure TPointData.SwapXY;

begin

  Tag := FX;

  FX := FY;

  FY := Tag;

end;

Notice the default specifier in the property declaration. This specifies the value in the run-time type information that the VCL streaming mechanism will compare the property value against before deciding whether to store it or not. It does not actually give the property a default value (a common misunderstanding) - that job is left to the programmer. In our case it’s easy; if the normal Create constructor is called, FX and FY will be zero anyway, as all object data fields are zeroed when the object is constructed.

We could have achieved the same net result as the default specifier, using the stored directive instead. The listing below is an alternative component definition for TPointData that uses a Boolean function specified with the stored directive, to decide whether to store the property in the stream or not. Note you can also use a Boolean data field or a Boolean constant.

  TPointData = class(TComponent)

  private

    FX, FY: Word;

    function IsX: Boolean;

    function IsY: Boolean;

  public

    constructor CreateXY(AOwner: TComponent; AX, AY: Word);

    procedure SwapXY;

  published

    property X: Word read FX write FX stored IsX;

    property Y: Word read FY write FY stored IsY;

  end;

...

function TPointData.IsX;

begin

  Result := X <> 0;

end;

function TPointData.IsY;

begin

  Result := Y <> 0;

end;

As was discussed briefly earlier, a stream has a method for writing a component’s properties, so we could change LoadBtnClick and SaveBtnClick as shown.

procedure TForm1.SaveBtnClick(Sender: TObject);

var

  Stream: TFileStream;

begin

  Stream := TFileStream.Create(DataFile, fmCreate);

  try

    for Loop := 0 to PointList.Count - 1 do

    begin

      Pt := TPointData(PointList.Items[Loop]);

      Stream.WriteComponent(Pt);

    end;

  finally

    Stream.Free;

  end;

  ClearPoints;

  PaintBox1.Invalidate;

end;

procedure TForm1.LoadBtnClick(Sender: TObject);

var

  Stream: TFileStream;

begin

  ClearPoints;

  Stream := TFileStream.Create(DataFile,(
    fmOpenRead or fmShareDenyWrite);

  try

    while Stream.Position <> Stream.Size do

    begin

      Pt := TPointData.Create;

      Stream.ReadComponent(Pt);

      PointList.Add(Pt);

    end;

  finally

    Stream.Free;

  end;

  Invalidate;

end;

But these versions have the same disadvantage as the old versions. We have to know which components are in the stream, and create them ourselves. The idea was to make the streaming mechanism construct the components as they are read from the stream. Fortunately, a simple change to the LoadBtnClick handler achieves this. The while loop changes to

    while Stream.Position <> Stream.Size do

      PointList.Add(Stream.ReadComponent(nil));

and now the stream (or more correctly the TReader it uses, where a TReader, like its compatriot the TWriter, are both descendants of TFiler) finds the name of the class in the stream and attempts to make an instance of the class. However, all that’s in the stream is a string, and that is not sufficient to make a new class with. It searches a list of classes it knows about (the class registration list, a private TList variable in the Classes unit called ClassList) to see if it can find a class reference whose name matches the string. If a match is found, then code for that class will exist in the EXE, but in this case it will be unsuccessful.

Normally when you place components on a form, as well as extra information going into the form file, the form class in the form unit has an object reference declaration added. This declaration goes in an unnamed section of the form class managed by Delphi, which happens to be the same as a published section. All the classes used in these object reference declarations are added into an internal field class table for that form which is used in addition to the global class registration list. Incidentally, the layout and positioning of this field class table can be gleaned from the ObjInfo.Pas unit described in the Object internals section earlier. A call to that unit’s VMTEntry(TForm1, vtFieldTable) would give you a pointer to a TFieldTable record, in which you would find a pointer to a TFieldClassTable.

For custom objects that are not already dealt with by a form, we need to register the classes. In the initialisation section of the form unit I have added

  RegisterClass(TPointData);

If you needed to register several classes, you could use

  RegisterClasses([TPointData, SecondClass, ThirdClass]);

The finished project is STRM2.DPR, but it is not yet done with. If you compare a stream written by STRM1.EXE and compare it with one written by STRM2.EXE you will see that the latter’s is considerably larger. All that is stored in the first version’s file are numbers. In the latest version we have information to allow the streaming mechanism to construct new objects. Close examination of this information shows that for each object in the stream there is a signature string, TPF0 (Turbo Pascal Filer version 0), a class name and some binary data representing the property values. There are also symbols indicating the size of each particular value (these are the ordinal values of some members of the Classes unit enumerated type TValueType, vaInt8 and vaInt16).

Using a container component (version 3)

Every component that gets explicitly written out is called a root component. Each root component is preceded by the signature TPF0, so that some simple validation can be performed on streams. The only way to avoid having this written for each point object would be to write all the points out at the same time, instead of iterating over all of them. We are unable to write the list out as it is based on type TObject, not TPersistent (or TComponent) and anyway it knows nothing about the items it maintains references to, other than their addresses.

Instead, for the next version we will replace the TList version of PointList with a component to keep hold of the points. There is no need to worry about adding list functionality - a component already has as much as we need, so TPointList can be based on type TComponent. When a component becomes the owner of another component, the owned component goes into an array property called Components, and another property called ComponentCount gets incremented.

There are several benefits to doing this. Since the list is a component, we can get it owned by the form, thus relinquishing our responsibility for destroying it. When we wish to write all our points out, providing we make the list own the points, we can simply write the list component out, and similar for reading (though see The Delphi Impact later). This means only one filer signature will be written and read. Also, the ClearPoints method, which empties the list and destroys all the points, can now become a simple call to the PointList’s DestroyComponents method, which will destroy all that it owns, i.e. all the points. The TPointList class in all its glory, along with some of the methods from STRM3.DPR are given here. Indeed, there was no real need to make a new class - TComponent would have been fine on its own.

TPointList = class(TComponent)

end;

...

procedure TForm1.PaintBox1Paint(Sender: TObject);

begin

  for Loop := 0 to PointList.ComponentCount - 1 do

  begin

    Pt := PointList.Components[Loop] as TPointData;

    if Loop = 0 then

      PaintBox1.Canvas.MoveTo(Pt.X, Pt.Y)

    else

      PaintBox1.Canvas.LineTo(Pt.X, Pt.Y)

  end;

end;

...

procedure TForm1.SaveBtnClick(Sender: TObject);

var

  Stream: TFileStream;

begin

  Stream := TFileStream.Create(DataFile, fmCreate);

  try

    Stream.WriteComponent(PointList);

  finally

    Stream.Free;

  end;

  ClearPoints;

  PaintBox1.Invalidate;

end;

...

initialization

  Randomize;

  RegisterClass(TPointData);

end.

Making properties with DefineProperties (Version 4)

There’s one other thing to mention about standard component streaming, and it again relates back to type TPersistent. A TWriter object will write the class name, instance name and properties of a component out to a stream. Sometimes it is not desirable to turn everything that should be streamed into a property. If this is the case, then there is another option available to allow storage of your data. TPersistent has a virtual method called DefineProperties. After all the real properties have been written, a TWriter calls a TPersistent derivative’s DefineProperties method which allows it to define fake properties, and dictate how they are written out.

To demonstrate, STRM4.DPR has had the TPointData properties removed, and X and Y are once again public data fields. DefineProperties has been overridden (see below) to define one fake property called XY. Inside a component’s DefineProperties method, you normally call the inherited method, but this one elects not to. In this case, the only property I want on the stream is my new XY property and nothing else.

  TPointData = class(TComponent)

  public

    X, Y: Word;

    constructor CreateXY(AOwner: TComponent; AX, AY: Word);

    procedure SwapXY;

    procedure DefineProperties(Filer: TFiler); override;

    procedure ReadData(Reader: TReader);

    procedure WriteData(Writer: TWriter);

  end;

...

procedure TPointData.DefineProperties(Filer: TFiler);

begin

  { Not calling inherited version cos I don't }

  { want any properties bar X and Y stored }

  Filer.DefineProperty('XY', ReadData, WriteData, X or Y <> 0);

end;

procedure TPointData.ReadData(Reader: TReader);

begin

  X := Reader.ReadInteger;

  Y := Reader.ReadInteger;

end;

procedure TPointData.WriteData(Writer: TWriter);

begin

  Writer.WriteInteger(X);

  Writer.WriteInteger(Y);

end;

To define a new property you call either DefineProperty or DefineBinaryProperty, each of which is a method of the passed in TFiler object. Normally the former is used, although graphic objects use the latter. DefineProperty takes a property name, a method that can read the property, a method to write the property and also a Boolean expression that dictates whether there is any data to store. The only combination of two values or-ed together that yield a zero are zero and zero, hence my expression is X or Y <> 0. As the listing shows, ReadInteger can be used to read a value of any integer type, and WriteInteger will write any integer type.

Saving a desktop file

We can take advantage of this component streaming ability to add a desktop-saving feature to our programs. The project DESKTOP.DPR does this. When the program exits, it saves the edit box and list box to a stream, which we can call our desktop file. When the program starts up, after all components have been read in and set up from the form, the overridden Loaded method reads the properties of the edit and list box back from the file. The net effect is each time you run the program, it looks just like it did when you last closed it, data and all. The Loaded method and the form’s OnClose event are below. Also included in that listing is the code for the Delete button. Delete removes the currently selected items from the list box, and caters for both single-selection and multiple-selection list boxes.

procedure TForm1.FormClose(Sender: TObject;(
  var Action: TCloseAction);

begin

  with TFileStream.Create(DataFile, fmCreate) do

    try

      WriteComponent(ItemEdt);

      WriteComponent(ItemsLst);

    finally

      Free;

    end;

end;

procedure TForm1.Loaded;

begin

  inherited Loaded;

  try

    with TFileStream.Create(DataFile, fmOpenRead or fmShareDenyWrite) do

      try

        ReadComponent(ItemEdt);

        ReadComponent(ItemsLst);

      finally

        Free;

      end;

  except

    { Smother desktop not found exception }

    { as it won’t be found on the first run }

    on EFOpenError do {nothing};

  end;

end;

procedure TForm1.DelBtnClick(Sender: TObject);

var

  Loop: Byte;

begin

  with ItemsLst, Items do

  begin

    BeginUpdate;

    if not MultiSelect then

      Delete(ItemIndex)

    else

      for Loop := Pred(Count) downto 0 do

        if Selected[Loop] then

          Delete(Loop);

    EndUpdate;

  end;

end;

Run-time resources (version 5)

Earlier, we saw how to make a DFM file, just like the ones Delphi generates. Indeed we can make a Windows resource file filled with components any time we like, but what’s the point? Well, going back to the point made some while ago, about object streams allowing you to remove initialisation from your program - it can be left to another program to set a stream up and your program can then read it. Let’s test out the theory.

Delphi generates an object resource file and your program reads in the resource at run-time. We will again amend the stream program, developed through this section, to do the same thing with some custom objects. First things first, we need a set-up program. SETUP.DPR on the CD shows how to generate a resource file with a list of five point objects in it. Because the file format of Windows resources changes between 16- and 32-bit, some conditional compilation is used to generate a uniquely named file. All the hard work is done in the Loaded method - called when the form has read itself in.

const

{$ifdef Windows}

  ResFile = 'Points.R16';

{$else}

  ResFile = 'Points.R32';

{$endif}

procedure TForm1.Loaded;

begin

  inherited Loaded;

  PointList := TPointList.Create(Self);

  Pt := TPointData.CreateXY(PointList, 10, 10);

  Pt := TPointData.CreateXY(PointList, 366, 10);

  Pt := TPointData.CreateXY(PointList, 366, 191);

  Pt := TPointData.CreateXY(PointList, 10, 191);

  Pt := TPointData.CreateXY(PointList, 10, 10);

  WriteComponentResFile(ResFile, PointList);

  MessageDlg('Job done!', mtInformation, [mbOk], 0);

  Application.Terminate;

end;

The five point objects draw out a square. The plan is that when the stream program starts up, it will read in its point list from a resource, rather than constructing one and waiting for the user to generate some random points. This means that as soon as the form shows on the screen, a square will be drawn on it.

The stream program will be an extension of STRM3.DPR - the one with the real properties, rather than faked ones. There are just a few changes to be made. Firstly, we need to link the resource into the program. A $R directive does that. Then we need to amend the initialisation section of the unit, so that both TPointData and TPointList are registered - we’re going to get the streaming mechanism to construct the list and the points. Lastly, we change the code in the form’s OnCreate handler. It currently creates a TPointList object. Instead we will call ReadComponentRes to read it in.

...

{$ifdef Windows}

  {$R Points.R16}

{$else}

  {$R Points.R32}

{$endif}

...

procedure TForm1.FormCreate(Sender: TObject);

begin

  PointList := ReadComponentRes(TPointList.ClassName, nil) as TPointList;

end;

...

initialization

  Randomize;

  RegisterClasses([TPointList, TPointData]);

end.

A rather pleasing side-effect of writing Delphi objects into resources is that the CONVERT.EXE program will now translate the binary file into a text file. A command-line of

 CONVERT POINTS.R16

yields the text shown below.

object TPointList

  object TPointData

    X = 10

    Y = 10

  end

  object TPointData

    X = 366

    Y = 10

  end

  object TPointData

    X = 366

    Y = 191

  end

  object TPointData

    X = 10

    Y = 191

  end

  object TPointData

    X = 10

    Y = 10

  end

end

Of course, like a form, this can be manually edited and changed back to a binary file with:

 CONVERT POINTS.TXT

 REN POINTS.DFM POINTS.R16

The Delphi 2 impact

And that should have been that. But as I found to my horror when I checked all these Delphi 1 examples in Delphi 2, shortly after its release, several of them didn’t work. STRM3, STRM4, STRM5 and SETUP failed to store any points in their streams. The common factor between these applications is that they all store the TPointList on the stream and expect all its owned components to be saved.

In Delphi 1 the TWriter object stores the root component’s owned components, providing they have no parent to do it for them. It finds that out by calling the HasParent function, which by default returns False. In Delphi 2 the TWriter object does not do this.

Drat! However, in both versions the TWriter does give a component an opportunity to write all (or some, or none of) the components it owns by calling one of the component’s methods. Unfortunately for us, the method, and approach to its calling differs between versions. Apparently, whilst adding the support for inherited forms, the streaming had a few architectural changes. Sounds like a job for conditional compilation to me. The version-proof way of writing a component, and getting all the owned components to be written to a stream is exemplified by the code shown next. All the failing examples mentioned above include the various changes.

  TPointList = class(TComponent)

  protected

{$ifdef Windows}

    procedure WriteComponents(Writer: TWriter); override;

{$else}

    procedure GetChildren(Proc: TGetChildProc); override;

{$endif}

  end;

  TPointData = class(TComponent)

  ...

{$ifdef Windows}

  protected

    function HasParent: Boolean; override;

{$endif}

  ...

  end;

...

{$ifdef Windows}

procedure TPointList.WriteComponents(Writer: TWriter);

var

  Loop: Integer;

begin

  { inherited version does nothing. Don’t call it }

  for Loop := 0 to ComponentCount - 1 do

    Writer.WriteComponent(Components[Loop]);

end;

{$else}

procedure TPointList.GetChildren(Proc: TGetChildProc);

var

  Loop: Integer;

begin

  { inherited version does nothing. Don’t call it }

  for Loop := 0 to ComponentCount - 1 do

    Proc(Components[Loop]);

end;

{$endif}

...

{$ifdef Windows}

function TPointData.HasParent: Boolean;

begin

  Result := True;

end;

{$endif}

...

Firstly, in 16-bit, the HasResult virtual method (of the owned component) needs overriding to return True, to suggest that the Delphi 1 TWriter does not write them out (we’d wind up with two copies of each TPointData object at the end otherwise).

Secondly, in 16-bit, the owner needs a WriteComponents method. This is passed a TWriter, and the component can call the writer’s WriteComponent method for any components that need streaming.

Lastly, in 32-bit, the owner needs a GetChildren method. This is passed a parameter of a procedural type, i.e. a reference to a TWriter method. For each component that needs streaming, you call that method and pass the component as a parameter.

This change from WriteComponents to GetChildren is a good idea. It provides a general iterater. You can iterate through all the components that the owner is interested in, and have any old routine of yours called for each one, with the component passed as a parameter. The only restriction is that your routine must be some procedure method that takes only one parameter - a component. An example project that shows this is ITERATE.DPR. Upon the push of Button1, Listbox1 is filled with information about each component on the form, including its name, class and ancestor. You can see that there is slightly less work to do in the Delphi 2 code.

procedure TForm1.AddToList(Child: TComponent);

begin

  with Child do

    Listbox1.Items.Add(Format('%s: %s = class(%s)',(
      [Name, ClassName, ClassParent.ClassName]));

end;

procedure TForm1.Button1Click(Sender: TObject);

var

  Loop: Integer;

begin

  Listbox1.Items.Clear;

{$ifdef Win32}

  GetChildren(AddToList);

{$else}

  for Loop := 0 to ComponentCount - 1 do

    AddToList(Components[Loop]);

{$endif}

end;

Sample file handling project

To do a bit of a recap of what has been covered so far on the subject of file handling, we will look at a reasonably simple program that uses a structured data file. To see the different options available for file handling, the program will be first written using a typed variable, then with an untyped file variable, once more with file handles and yet again with streams.

The program is not particularly adventurous, for simplicity of reading - it will allow storage of records of information consisting of a name and a date of birth. The file will not exist to start with, and so the program needs to know how to create it, and must obviously support adding records. It will also attempt to handle the various error situations which may arise, using the natural error mechanism of the file system used (i.e. exceptions for file variables).

Typed file variables (version 1)

The program (NAMES1.DPR) is split over two units. The first is the main form unit, NAMES1U.PAS, and implements the user interface. I won’t go into what it does in any detail, though I’ll mention that it spends some time ensuring that clipboard-oriented menu items and speed buttons are enabled and disabled when appropriate, and the same for navigational menu items and speed buttons. Additionally it implements the functionality behind those buttons/menus.

The important part of the code is left to the second unit (NAMES1U2.PAS), which has nothing to do with the user interface. This unit implements a class to represent a data file and also defines the data record. The plan is that only this second unit will need adjusting for the different file types. The TDataFile class interface will remain static, but the implementation will change.

type

  TDataRec = packed record

    { The form's edit box has its }

    { MaxLength property set to 30 }

    Name: String[30];

    { Only interested in the date portion }

    { of this date/time value }

    DOB: TDateTime;

  end;

  TDataFile = class

  private

    FDataFile: File of TDataRec;

  protected

    function GetCount: Longint;

    function GetCurrent: Longint;

    function GetRecord(Index: Longint): TDataRec;

    procedure SetCurrent(RecNo: Longint);

    procedure SetRecord(Index: Longint; const DataRec: TDataRec);

  public

    constructor Create;

    destructor Destroy; override;

    property Count: Longint read GetCount;

    property Current: Longint read GetCurrent write SetCurrent;

    property Records[Index: Longint]: TDataRec(
      read GetRecord write SetRecord; default;

  end;

There are seven principal methods of interest (mostly property access methods) which all deal with the file variable. Some are fairly simplistic, like GetCount which returns the number of record in the file and GetCurrent which reports the position in the file. You can see from the listing below that they are simple wrappers around FileSize and FilePos. The code in SetCurrent is almost only a call to Seek, but there is a check to ensure that the caller is not requesting a position past the end of the file.

function TDataFile.GetCount: Longint;

begin

  Result := FileSize(FDataFile);

end;

function TDataFile.GetCurrent: Longint;

begin

  Result := FilePos(FDataFile);

end;

procedure TDataFile.SetCurrent(RecNo: Longint);

begin

  { Anything past EOF is considered EOF }

  if RecNo > Count then

    RecNo := Count;

  Seek(FDataFile, RecNo);

end;

The more interesting methods are the constructor and destructor, and also the Records property access methods GetRecord and SetRecord.

constructor TDataFile.Create;

begin

  { Make current directory where }

  { EXE file is, just in case }

  ChDir(ExtractFileDir(Application.ExeName));

  AssignFile(FDataFile, FileName);

  FileMode := fmOpenReadWrite or fmShareDenyNone;

  try

    { Make file if it ain't there }

    if not FileExists(FileName) then

      Rewrite(FDataFile);

    Reset(FDataFile);

  except

    on E: EInOutError do

    begin

      { In case Rewrite succeeded but Reset failed }

      if TFileRec(FDataFile).Mode = fmInOut then

        CloseFile(FDataFile);

      { Customise the exception and re-raise it }

      E.Message := 'Failed to create or open ' + FileName;

      raise;

    end;

  end;

end;

The constructor changes to the directory where the program’s EXE file resides and sets FileMode up for normal sharing. Notice that it uses ExtractFileDir rather than ExtractFilePath - the latter leaves backslash characters on paths and ChDir doesn’t like that. ExtractFileDir is better but is new in Delphi 2 so the unit codes up a replacement for Delphi 1 users.

{$ifdef Ver80}

function ExtractFileDir(const FileName: String): String;

var

  I: Integer;

begin

  Result := ExtractFilePath(FileName);

  I := Length(Result);

  if (I > 1) and (FileName[I] = '\') and (FileName[I - 1] <> ':') then

    { This is compiled in Delphi 1 only, so this is fine }

    Dec(Result[0]);

end;

{$endif}

If there is no data file found an attempt is made to make one, followed by the normal call to Reset which should open the file in sharing mode. If there is a problem, for example another program already has the file open in a mode that excludes us from writing, then an exception is raised and caught. Before re-raising the error with a custom message, code executes to ensure the file is definitely shut taking advantage of the file variable structure record TFileRec and one of the file open mode constants, fmInOut.

destructor TDataFile.Destroy;

begin

  if TFileRec(FDataFile).Mode = fmInOut then

    CloseFile(FDataFile);

  inherited Destroy;

end;

The destructor ensures the file is closed before letting the object destroy itself. If you try and close a file that is not open, you get an exception. Some would say that these days you should be in the mindset of exception handling, and just close the file, but trap an exception if it occurs. However doing it this way shows a use for the Mode field in the TFileRec structure.

function TDataFile.GetRecord(Index: Longint): TDataRec;

begin

  try

    Current := Index;

    Read(FDataFile, Result);

    { Go back to the beginning of the read record }

    Current := Index;

  except

    raise EListError.CreateRes(SListIndexError);

  end;

end;

procedure TDataFile.SetRecord(Index: Longint; const DataRec: TDataRec);

var

  X: EInOutError;

begin

  Current := Index;

  if not LockFileVar(FDataFile, Current, True) then

  begin

    X := EInOutError.Create('Cannot lock file');

    { Set up a file access denied type exception }

    X.ErrorCode := 5;

    raise X;

  end;

  try

    { DataRec is passed as a const (pass by }

    { reference, but not allowed to be treated/ }

    { passed as a var parameter). }

    { We can get around this by dereferencing its }

    { address with an appropriate typecast }

    Write(FDataFile, TDataRec((@DataRec)^));

    { Go back to the start of the written record }

    Current := Index;

  finally

    LockFileVar(FDataFile, Current, False);

  end;

end;

GetRecord retrieves a specified record by positioning the file pointer and using Read. If Read yields any kind of problem an Index out of bounds exception is raised to suggest that maybe an invalid record was requested. SetRecord is less trivial - it tries to lock a record in the file so it can exclusively write a record without fear of conflict with anyone else. If the lock cannot be placed an I/O exception is raised, otherwise the record is written and the lock is removed.

The point of the TDataRec((@DataRec)^) element in the call to write is simply to show how you can get around the normal restrictions on const parameters - i.e. that they can’t be passed as var parameters. Normally I wouldn’t write such an expression. It would be clearer to declare a local variable of type TDataRec, assign DataRec to it and pass that instead.

Untyped file variables (version 2)

When rewriting this for untyped files (see project NAMES2.DPR on the CD), there are very few changes to make (four in total). First the definition of FDataFile needs to change to be:

FDataFile: File;

To give the untyped file a record size, we modify the call to Reset in the constructor:

Reset(FDataFile, SizeOf(TDataRec));

Finally, GetRecord and SetRecord must use BlockRead and BlockWrite instead of Read and Write:

BlockRead(FDataFile, Result, 1, Count);

...

BlockWrite(FDataFile, TDataRec((@DataRec)^), 1, Count);

The fourth parameter is an optional Word variable which is used to detect problems. It returns the number of records (since we set a record size) read or written. If it is less than we requested, i.e. if it is zero, then there was a file error.

File handles (version 3)

The version for file handles (NAMES3.DPR) has a few more involved changes in NAMES3U2.PAS. FDataFile is now defined as an Integer. The constructor contains no exception handling as file handle routines don’t generate exceptions, although it does generate some exceptions if there is a problem.

In order to cater for the lack of a FileSize routine I implemented one using the logic described earlier. This isn’t necessary in Win32, since it has its own GetFileSize routine. There is conditional compilation to call the relevant one in GetCount. GetRecord and SetRecord are modified to use FileRead and FileWrite, but also SetRecord uses LockFileArea instead of LockFileVar. As an example of the changes, SetCurrent is listed here.

procedure TDataFile.SetCurrent(RecNo: Longint);

begin

  { Anything past EOF is considered EOF }

  if RecNo > Count then

    RecNo := Count;

  FileSeek(FDataFile, RecNo * SizeOf(TDataRec), soFromBeginning);

end;

Streams (version 4)

The “database” program changes now in Names4.Dpr to use a TFileStream. The TDataFile constructor has the job of either opening the data file, or creating and then opening the data file. The code with a file stream looks like this.

constructor TDataFile.Create;

begin

  { Make current directory where EXE file is, just in case }

  ChDir(ExtractFileDir(Application.ExeName));

  { Make file if it ain't there }

  if not FileExists(FileName) then

  begin

    { We don't need a try..finally block here cos if the }

    { file creation fails, the constructor raises an }

    { exception which causes the object to be freed }

    FDataFile := TFileStream.Create(FileName, fmCreate);

    FDataFile.Free;

  end;

  FDataFile := TFileStream.Create(FileName,(
    fmOpenReadWrite or fmShareDenyNone);

end;

and the destructor simply calls FDataFile.Free. The size of the file in terms of data records, as returned by GetCount can be implemented as:

  Result := FDataFile.Size div SizeOf(TDataRec);

We can find the current position in GetCurrent by using Seek, as we did last time around:

  Result := FDataFile.Seek(0, soFromCurrent);

  if Result > -1 then

    Result := Result div SizeOf(TDataRec);

and go to a requested record similarly:

FDataFile.Seek(RecNo * SizeOf(TDataRec), soFromBeginning);

The record reading and writing operations are also pretty much the same as before. Refer to the Names4U2.Pas unit on the CD for the full implementation.

Text File Device Drivers

Have you ever used the WinCrt unit in Delphi 1? Or the console mode facility in Delphi 2? Or the AssignPrn procedure from the Printers unit? In that case you have used a text file device driver (TFDD), although you might not have known it at the time. You have accessed these input and output facilities by using Read/ReadLn and/or Write/WriteLn and the information has come from the keyboard, or gone to the screen in some fashion, or to the printer.

These three TFDDs are accessed in two slightly different ways. The printer support is used just like a normal text file variable, but with AssignPrn being called instead of AssignFile:

var

  T: TextFile;

...

AssignPrn(T);

WriteLn(T, ‘Hello world‘);

However the WinCrt and console mode support is accessed specifying no file variable, e.g.:

var

  S: String;

...

WriteLn(‘Hello world‘);

ReadLn(S);

Read/Ln and Write/Ln always operate on text files despite this evidence to the contrary. As it turns out, when no file variable is specified two System unit text file variables, Input and Output, are implicitly used. The above is exactly the same as writing this:

var

  S: String;

...

WriteLn(Output, ‘Hello world‘);

ReadLn(Input, S);

Using this information, you could access the printer without specifying a file variable as follows:

AssignPrn(Output);

WriteLn(‘Hello world‘);

The idea of a TFDD is to allow information to be gathered from, or sent to, some “device” though the Read/Ln and Write/Ln standard procedures via a text file variable, be it an implicitly or explicitly declared one. The nature of the device does not matter, so long as the relevant supporting code to talk to it is present in the TFDD’s implementation. In the cases described above, the devices included the keyboard, a Windows GUI window, a Windows console window and a printer.

Bearing in mind that a text file can only be accessed in read-only mode (using Reset or Append) or write-only mode (using Rewrite), the TFDD can be written to allow any given any file variable read-only or write-only access to the device.

Writing a text file device driver

When implementing a TFDD you need to write functions for five purposes. One of your functions will be called when the device is opened (using Reset, Rewrite or Append), when the device is read from (using Read/Ln), written to (Write/Ln), flushed (implicitly performed after input and output) and closed (CloseFile). The way you set these functions up is by writing a customised AssignFile substitute which assigns the various functions to four function pointers in the text file variable. Even though there are five possible functions, two are mutually exclusive - the reading and writing routines are never used at the same time, so they both get assigned to the same function pointer. called InOutFunc.

Note that the real AssignFile sets up functions in the RTL to deal with opening/reading/writing etc. files - normal text file access is achieved also using the TFDD mechanism where the “device” is a text file on a disk.

TTextRec record type

In order to get access to the function pointers in the text file variable, the customised AssignFile procedure must typecast it into a TTextRec record. This type, defined in the SysUtils unit, defines the internal layout of a TextFile variable. The definition differs slightly between Delphi 1 and 2, though not with any important consequences.

Delphi 1 TTextRec
Delphi 2 TTextRec

PTextBuf = ^TTextBuf;

TTextBuf = array[0..127] of Char;

TTextRec = record

  Handle: Word;

  Mode: Word;

  BufSize: Word;

  Private: Word;

  BufPos: Word;

  BufEnd: Word;

  BufPtr: PTextBuf;

  OpenFunc: Pointer;

  InOutFunc: Pointer;

  FlushFunc: Pointer;

  CloseFunc: Pointer;

  UserData: array[1..16] of Byte;

  Name: array[0..79] of Char;

  Buffer: TTextBuf;

end;
PTextBuf = ^TTextBuf;

TTextBuf = array[0..127] of Char;

TTextRec = record

  Handle: Integer;

  Mode: Integer;

  BufSize: Cardinal;

  { Note no Private field }

  BufPos: Cardinal;

  BufEnd: Cardinal;

  BufPtr: PChar;

  OpenFunc: Pointer;

  InOutFunc: Pointer;

  FlushFunc: Pointer;

  CloseFunc: Pointer;

  UserData: array[1..32] of Byte;

  Name: array[0..259] of Char;

  Buffer: TTextBuf;

end;

The Handle data field normally stores the file handle of the represented text file and the Name field holds its file name. Mode tells you the current state of the file. For a text file this can be fmClosed, fmInput or fmOuput - the other possible value, fmInOut doesn’t apply to text files. The Buffer field is the initial text buffer, 128 bytes in size. This buffer is pointed to by BufPtr, although a different buffer can be set up with a call to SetTextBuf. BufSize tells how big the buffer is, BufPos says how far through the data in the buffer we are, and BufEnd signifies the end of the valid data in the buffer. UserData is some space unused by the Delphi file system, intended for use by the TFDD writer for storing private information. In Delphi 1, Private is not used and so can be ignored, which leaves the four pointer fields.

Here is a list of the responsibilities of the customised AssignFile and also for each of the functions referenced by these pointer fields.

AssignFile substitute procedure

This should associate all four of the device-interface functions to the function pointer fields in a text file variable. However this guideline is often not followed to the letter. Since the function referred to by the OpenFunc function will always be called before any of the others, many AssignFile routines only set up OpenFunc. They leave all the other pointers for OpenFunc to set up. Depending on the value of Mode, OpenFunc can set up the remaining pointers to either input- or output-based routines. This can save the InOutFunc, FlushFunc and CloseFunc functions from having to check the Mode value.

As well as the function pointers, this must also assign fmClosed to Mode, store the size of whatever buffer is being used in BufSize, place the address of the buffer in BufPtr, and clear the Name null-terminated string field. It can also do any other initialisation required, such as storing some data in UserData. One other thing that is sometimes done is to set the Handle field (which won’t be used in a TFDD) to $FFFF.

OpenFunc function

This is called when the file is opened by Reset, Rewrite or Append. Upon entry, the Mode field contains fmInput, fmOutput or fmInOut respectively as an indicator as to which procedure was called. Any preparation required for input or output, as indicated by Mode, can be done. If Append was called, and Mode therefore has a value of fmInOut, it must be changed to fmOutput.

InOutFunc function

This function gets invoked by Read, ReadLn, Write, WriteLn, Eof, Eoln, SeekEof, SeekEoln, Close and Flush (if Mode is fmOutput) when device input or output is needed.

If Mode is fmInput, this routine must read up to BufSize characters into BufPtr^, assign to BufEnd the number of characters read and set BufPos to zero. If BufEnd is zero, Eof will be True for the file.

If Mode is fmOutput, this routine should write BufPos characters from BufPtr^ and set BufPos to zero.

FlushFunc function

This routine is called by default at the end of each Read, ReadLn, Write and WriteLn. This function can optionally flush the text-file buffer. Note that this is curiously not called by the Flush procedure, which instead calls InOutFunc if Mode is fmOutput.

If Mode is fmInput, this can set BufPos and BufEnd to zero to flush the remaining unread characters in the buffer. This feature is hardly ever used.

If Mode is fmOutput, this can behave much like the InOutFunc function, thus ensuring that text written to the file variable gets sent to the device immediately. If this routine does nothing, the text won’t appear in the device until the buffer becomes full, the file is closed, or the Flush procedure is called. It is because this routine can optionally not do anything that Flush does not use it.

CloseFunc function

CloseFile invokes this function to when closing a text file associated with a device. Additionally, Reset, Rewrite and Append will call this if the file is already open. If Mode is fmOutput, then InOutFunc will be called before CloseFunc to ensure all data is written to the device.

Errors

Each of these four or five functions (not all of which are necessarily required) have the following interface:

function DeviceFunc(var F: TTextRec): Integer;

In Delphi 1, they must be compiled in the far call model. This can be done by, amongst other things, declaring them in the interface section of a unit, or by placing the far keyword after the declaration line. 

If the functions return a non-zero value, then it signifies an I/O error. The return value of each function become the value that IOResult will return, or that may turn up as the ErrorCode property of an EInOutError exception, depending on the state of the I/O checking option.

An input/output device

Our first look at implementing a TFDD will involve an edit control. The project TFDD1.DPR does this, and the TFDD is in RWEditU.PAS. In this case, we are keeping track of which edit control is associated with which file variable by storing the edit’s object reference in the UserData field. To do this requires a record to be defined to use in typecasting UserData, which is not defined of an appropriate type. Here is a variant record that fits the bill - remember that UserData doubles in size in Delphi 2.

const

{$ifdef Win32}

  FillerMax = 32;

{$else}

  FillerMax = 16;

{$endif}

type

  TUserData = packed record

    case Byte of

      1: (

        Edit: TEdit);

      2: (

        Filler: array[1..FillerMax] of Byte);

  end;

Note that a non-variant record like this would have sufficed just as well:

  TUserData = packed record

    Edit: TEdit;

    Filler: array[SizeOf(TEdit)+1..FillerMax] of Byte;

  end;

The device interface routines are forward declared at the top of the unit’s implementation section.

function RWEditOpen(var F: TTextRec): Integer; far; forward;

function RWEditInput(var F: TTextRec): Integer; far; forward;

function RWEditOutput(var F: TTextRec): Integer; far; forward;

function RWEditClose(var F: TTextRec): Integer; far; forward;

The customised AssignFile routine does its housekeeping as shown below.

procedure AssignRWEdit(var F: TextFile; E: TEdit);

begin

  { Set up text file variable }

  with TTextRec(F) do

  begin

    Handle := $FFFF;

    OpenFunc := @RWEditOpen;

    Mode := fmClosed;

    BufSize := SizeOf(Buffer);

    BufPtr := @Buffer;

    Name[0] := #0;

    { Set up edit control - store it in the text file variable }

    TUserData(UserData).Edit := E;

  end;

end;

The OpenFunc routine does all the function pointer assignments based on the Mode value, as described above. You’ll notice that the FlushFunc pointer is set to nil for input, but to the InOutFunc value for output. The CloseFunc on the other hand, has a very easy ride of it.

function RWEditOpen(var F: TTextRec): Integer;

begin

  Result := 0;

  with F do

  begin

    if Mode = fmInput then

    begin

      InOutFunc := @RWEditInput;

      FlushFunc := nil;

    end

    else

    begin

      Mode := fmOutput;

      InOutFunc := @RWEditOutput;

      FlushFunc := @RWEditOutput;

    end;

    CloseFunc := @RWEditClose;

  end;

end;

function RWEditClose(var F: TTextRec): Integer;

begin

  Result := 0;

end;

The input and output routines follow the previously outlined guidelines to read from or write to the edit control. Note that RWEditOutput checks that the program isn’t terminating. Remember that CloseFile will call the InOutFunc routine first. This is important since the shutdown code in a Delphi 2 application calls CloseFile for Output. If we continue trying to access the edit control, we are risking an access violation.

function RWEditInput(var F: TTextRec): Integer;

begin

  Result := 0;

  with F, TUserData(UserData).Edit do

  begin

    BufPos := 0;

    BufEnd := GetTextBuf(PChar(BufPtr), BufSize);

    { Pop a carriage return line feed combo in }

    StrCat(PChar(BufPtr), #13#10);

    Inc(BufEnd, 2);

    { Clear the edit }

    Text := '';

  end;

end;

function RWEditOutput(var F: TTextRec): Integer;

var

  { Temporary PChar holder }

  Buf: packed array[0..255] of Char;

begin

  Result := 0;

  { This gets called when a Delphi 2 app shuts, in closing Output }

  { Since it refers to the edit which won't exist, don't run it }

  if not Application.Terminated then

    with F, TUserData(UserData).Edit do

      if BufPos <> 0 then

      begin

        { Get PChar with BufPos characters in }

        StrLCopy(Buf, PChar(BufPtr), BufPos);

        { Put that in the edit }

        SetTextBuf(Buf);

        { Reset BufPos }

        BufPos := 0;

      end;

end;

The form unit in the TFDD1.DPR project, TFDDU.PAS, sets up the TFDD in the form’s OnCreate handler.

{ Can't do this in the TFDD unit initialisation as it relies }

{ on an object in the form that only gets created after }

{ initialisation sections have finished executing }

AssignRWEdit(Input, Edit1);

Reset(Input);

AssignRWEdit(Output, Edit1);

Rewrite(Output);

This allows other event handlers to use Read/ReadLn and Write/WriteLn without specifying a file variable - Input and Output will be used implicitly. There are two buttons and an edit on the form. The first button reads two floating point values from the edit and then writes some text back in, where the second one reads a string from the edit. The code below from the two buttons’ OnClick handlers show how these I/O procedures can liaise with the edit.

procedure TForm1.Button1Click(Sender: TObject);

var

  D1, D2: Double;

begin

{$ifdef OnePossibility}

  ReadLn(D1, D2);

{$else}

  Read(D1);

  Read(D2);

  ReadLn;

{$endif}

  ShowMessage(Format('First value: %f', [D1]));

  ShowMessage(Format('Second value: %f', [D2]));

  Write('Type something in me and push the 2nd button');

end;

procedure TForm1.Button2Click(Sender: TObject);

var

  S: String;

begin

  ReadLn(S);

  ShowMessage(Format('You wrote: %s', [S]));

  Write('Game over');

end;

An output only debugging device

This time, for a second example, we will look at a write-only TFDD. This one, demonstrated by TFDD2.DPR, will act as a debugging tracing type tool - the strings you give to WriteLn are displayed in a separate window. When you come to deploy an application using this TFDD, you can modify it slightly using a compiler directive so no output occurs.

Since this TFDD relies on no external edit controls or other objects it can set itself up (i.e. call its AssignFile substitute on the Output file variable) in its unit initialisation section. The following snippet comes from DebugU.Pas.

initialization

  AssignDebug(Output);

  Rewrite(Output);

end.

The AssignDebug procedure does much the same as the earlier AssignRWEdit did, leaving most of the TFDD function pointers to be set up by the OpenFunc routine.

procedure AssignDebug(var F: TextFile);

begin

  { Set up text file variable }

  with TTextRec(F) do

  begin

    Handle := $FFFF;

    OpenFunc := @DebugOpen;

    Mode := fmClosed;

    BufSize := SizeOf(Buffer);

    BufPtr := @Buffer;

    Name[0] := #0;

  end;

end;

DebugOpen ensures the device is only opened for output by causing an I/O error 5 (access denied) if opened with Reset. Otherwise it sets up the pointers and creates the debug output form (under control of conditional compilation). DebugClose will similarly free the form.

function DebugOpen(var F: TTextRec): Integer;

begin

  Result := 0;

  with F do

  begin

    if Mode = fmInput then

      { Access denied }

      Result := 5

    else

    begin

      Mode := fmOutput;

      InOutFunc := @DebugOutput;

      FlushFunc := @DebugOutput;

    end;

    CloseFunc := @DebugClose;

  end;

{$ifdef Debugging}

  DebugFrm := TDebugFrm.Create(Application);

{$endif}

end;

function DebugClose(var F: TTextRec): Integer;

begin

  Result := 0;

{$ifdef Debugging}

  DebugFrm.Free;

{$endif}

end;

The output routine ensures the form is visible. This means that the first WriteLn will cause the debugging form to appear on the desktop, as it starts off hidden. It then sets the selected text in the form’s memo to be the supplied text. Normally, there will be no selected text, and the caret will be at the end of the memo’s text, meaning the new text will be appended onto the memo.

function DebugOutput(var F: TTextRec): Integer;

var

  Buf: array[0..255] of Char;

begin

  Result := 0;

{$ifdef Debugging}

  { This gets called when a Delphi 2 app shuts, in closing Output }

  { Since it refers to the form which won't exist, don't run it }

  if not Application.Terminated then

  begin

    { If output form ain't showing, show it }

    if not DebugFrm.Visible then

      DebugFrm.Show;

    with F do

    begin

      StrLCopy(Buf, PChar(BufPtr), BufPos);

      DebugFrm.DebugMemo.SelText :=

        StrPas(Buf);

      BufPos := 0;

    end;

  end;

{$endif}

end;

Notice again that conditional compilation is used to prevent references to the form. In fact the entire debugging form class is conditionally compiled, as is the compiler directive that links the form into the executable. There is only one event handler in the form, for OnClose, which prevents it from being closed, apart from by normal application termination.

{$ifdef Debugging}

procedure TDebugFrm.FormClose(Sender: TObject; var Action: TCloseAction);

begin

  Action := caNone;

end;

{$endif}

The form that uses this TFDD has a button, whose OnClick handler writes a message to the debug window, and an edit whose OnChange handler keeps updating the debug window with the current text in the edit. One of these is listed below.

procedure TForm1.Edit1Change(Sender: TObject);

begin

  WriteLn('Edit says: ' + Edit1.Text);

end;

In order to get a deployable application, with no TFDD form resource included, simply remove the compiler directive from the top of the unit. It currently defines the symbol Debugging. Deleting the line will stop it being defined.

TFDD References

Borland Pascal With Objects 7.0 Language Guide, Chapter 14, Input and Output.

Delphi 1 Client/Server source files ...\SOURCE\RTL\WIN\WINCRT.PAS
Delphi 1 or 2 Client/Server source files ...\SOURCE\VCL\PRINTERS.PAS

